Theories and Data Structures

Jacques Carette, Musa Al-hassy, Wolfram Kahl

June 14, 2017

Abstract

We aim to show how common data-structures naturally arise from elementary mathematical theories.

In particular, we answer the following questions:

- Why do lists pop-up more frequently to the average programmer than, say, their duals: bags?
- More simply, why do unit and empty types occur so naturally? What about enumerations/sums and records/products?
- Why is it that dependent sums and products do not pop-up expicitly to the average programmer? They arise naturally all the time as tuples and as classes.
- How do we get the usual toolbox of functions and helpful combinators for a particular data type? Are they "built into" the type?
- Is it that the average programmer works in the category of classical Sets, with functions and propositional equality? Does this result in some "free constructions" not easily made computable since mathematicians usually work in the category of Setoids but tend to quotient to arrive in Sets? —where quotienting is not computably feasible, in Sets at-least; and why is that?

$\boxed{\textbf{???}}$

This research is supported by the National Science and Engineering Research Council (NSERC), Canada

Contents

1 Introduction

???

2 Overview

???

The Agda source code for this development is available on-line at the following URL:

`https://github.com/JacquesCarette/TheoriesAndDataStructures`

3 Obtaining Forgetful Functors

We aim to realise a "toolkit" for an data-structure by considering a free construction and proving it adjoint to a forgetful functor. Since the majority of our theories are built on the category Set, we begin my making a helper method to produce the forgetful functors from as little information as needed about the mathematical structure being studied.

Indeed, it is a common scenario where we have an algebraic structure with a single carrier set and we are interested in the categories of such structures along with functions preserving the structure.

We consider a type of "algebras" built upon the category of Sets —in that, every algebra has a carrier set and every homomorphism is a essentially a function between carrier sets where the composition of homomorphisms is essentially the composition of functions and the identity homomorphism is essentially the identity function.

Such algebras consistute a category from which we obtain a method to Forgetful functor builder for single-sorted algebras to Sets.

```
module Forget where
open import Level
open import Categories.Category using (Category)
open import Categories.Functor  using (Functor)
open import Categories.Agda     using (Sets)
open import Function2
open import Function
open import EqualityCombinators
```

[MA: *For one reason or another, the module head is not making the imports smaller.* **]**

A OneSortedAlg is essentially the details of a forgetful functor from some category to Sets,

```
record OneSortedAlg (ℓ : Level) : Set (suc (suc ℓ)) where
  field
    Alg        : Set (suc ℓ)
    Carrier    : Alg → Set ℓ
    Hom        : Alg → Alg → Set ℓ
    mor        : {A B : Alg} → Hom A B → (Carrier A → Carrier B)
    comp       : {A B C : Alg} → Hom B C → Hom A B → Hom A C
    .comp-is-∘ : {A B C : Alg} {g : Hom B C} {f : Hom A B} → mor (comp g f) ≐ mor g ∘ mor f
    Id         : {A : Alg} → Hom A A
    .Id-is-id  : {A : Alg} → mor (Id {A}) ≐ id
```

The aforementioned claim that algebras and their structure preserving morphisms form a category can be realised due to the coherency conditions we requested viz the morphism operation on homomorphisms is functorial.

```
open import Relation.Binary.SetoidReasoning
oneSortedCategory : (ℓ : Level) → OneSortedAlg ℓ → Category (suc ℓ) ℓ ℓ
oneSortedCategory ℓ A = record
  {Obj   = Alg
  ; _⇒_  = Hom
  ; _≡_   = λ F G → mor F ≐ mor G
  ; id     = Id
  ; _∘_  = comp
  ; assoc = λ {A B C D} {F} {G} {H} → begin⟨ ≐-setoid (Carrier A) (Carrier D) ⟩
    mor (comp (comp H G) F) ≈⟨ comp-is-∘                        ⟩
    mor (comp H G) ∘ mor F   ≈⟨ ∘-≐-cong₁ _ comp-is-∘           ⟩
    mor H ∘ mor G ∘ mor F     ≈˘⟨ ∘-≐-cong₂ (mor H) comp-is-∘ ⟩
    mor H ∘ mor (comp G F)   ≈˘⟨ comp-is-∘                       ⟩
    mor (comp H (comp G F)) ■
  ; identityˡ = λ {{f = f} → comp-is-∘ ⟨≐≐⟩ Id-is-id ∘ mor f}
  ; identityʳ = λ {{f = f} → comp-is-∘ ⟨≐≐⟩ ≡.cong (mor f) ∘ Id-is-id}
  ; equiv      = record {IsEquivalence ≐-isEquivalence}
  ; ∘-resp-≡  = λ f≈h g≈k → comp-is-∘ ⟨≐≐⟩ ∘-resp-≐ f≈h g≈k ⟨≐≐⟩ ≐-sym comp-is-∘
  }
  where open OneSortedAlg A; open import Relation.Binary using (IsEquivalence)
```

The fact that the algebras are built on the category of sets is captured by the existence of a forgetful functor.

```
mkForgetful : (ℓ : Level) (A : OneSortedAlg ℓ) → Functor (oneSortedCategory ℓ A) (Sets ℓ)
mkForgetful ℓ A = record
  {F₀              = Carrier
  ; F₁              = mor
  ; identity        = Id-is-id $ᵢ
  ; homomorphism = comp-is-∘ $ᵢ
  ; F-resp-≡       = _$ᵢ
  }
  where open OneSortedAlg A
```

That is, the constituents of a OneSortedAlgebra suffice to produce a category and a so-called presheaf as well.

4 Equality Combinators

Here we export all equality related concepts, including those for propositional and function extensional equality.

```
module EqualityCombinators where
open import Level
```

4.1 Propositional Equality

We use one of Agda's features to qualify all propositional equality properties by "≡." for the sake of clarity and to avoid name clashes with similar other properties.

```
import Relation.Binary.PropositionalEquality
module ≡ = Relation.Binary.PropositionalEquality
open ≡ using (_≡_) public
```

We also provide two handy-dandy combinators for common uses of transitivity proofs.

```
_⟨≡≡⟩_ = ≡.trans
_⟨≡≡˘⟩_ : {a : Level} {A : Set a} {x y z : A} → x ≡ y → z ≡ y → x ≡ z
x≈y ⟨≡≡˘⟩ z≈y = x≈y ⟨≡≡⟩ ≡.sym z≈y
```

4.2 Function Extensionality

We bring into scope pointwise equality, _≐_, and provide a proof that it constitutes an equivalence relation —where the source and target of the functions being compared are left implicit.

```
open ≡ using () renaming (_→-setoid_ to ≐-setoid; _≗_ to _≐_) public
open import Relation.Binary using (IsEquivalence; Setoid)
module _ {a b : Level} {A : Set a} {B : Set b} where
  ≐-isEquivalence : IsEquivalence (_≐_ {A = A} {B})
  ≐-isEquivalence = record {Setoid (≐-setoid A B)}
  open IsEquivalence ≐-isEquivalence public
    renaming (refl to ≐-refl; sym to ≐-sym; trans to ≐-trans)
  open import Equiv public using (∘-resp-≐)   -- To do: port this over here!
    renaming (cong∘l to ∘-≐-cong₂; cong∘r to ∘-≐-cong₁)
infixr 5 _⟨≐≐⟩_
_⟨≐≐⟩_ = ≐-trans
```

Note that the precedence of this last operator is lower than that of function composition so as to avoid superfluous parenthesis.

4.3 Equiv

We form some combinators for HoTT like reasoning.

```
cong₂D : ∀ {a b c} {A : Set a} {B : A → Set b} {C : Set c}
  (f : (x : A) → B x → C)
  → {x₁ x₂ : A} {y₁ : B x₁} {y₂ : B x₂}
  → (x₂≡x₁ : x₂ ≡ x₁) → ≡.subst B x₂≡x₁ y₂ ≡ y₁ → f x₁ y₁ ≡ f x₂ y₂
cong₂D f ≡.refl ≡.refl = ≡.refl
open import Equiv public using (_≃_; id≃; sym≃; trans≃; qinv)
infix 3 _□
infixr 2 _≃⟨_⟩_
_≃⟨_⟩_ : {x y z : Level} (X : Set x) {Y : Set y} {Z : Set z}
  → X ≃ Y → Y ≃ Z → X ≃ Z
X ≃⟨ X≃Y ⟩ Y≃Z = trans≃ X≃Y Y≃Z
_□ : {x : Level} (X : Set x) → X ≃ X
X □ = id≃
```

[MA: *Consider moving pertinent material here from Equiv.lagda at the end.* **]**

4.4 Making symmetry calls less intrusive

It is common that we want to use an equality within a calculation as a right-to-left rewrite rule which is accomplished by utilizing its symmetry property. We simplify this rendition, thereby saving an explicit call and parenthesis in-favour of a less hinder-some notation.

Among other places, I want to use this combinator in module Forget's proof of associativity for oneSortedCategory

```
module _ {c l : Level} {S : Setoid c l} where
  open import Relation.Binary.SetoidReasoning using (_≈⟨_⟩_)
  open import Relation.Binary.EqReasoning using (_IsRelatedTo_)
  open Setoid S
  infixr 2 _≈˘⟨_⟩_
  _≈˘⟨_⟩_ : ∀ (x {y z} : Carrier) → y ≈ x → _IsRelatedTo_ S y z → _IsRelatedTo_ S x z
  x ≈˘⟨ y≈x ⟩ y≈z = x ≈⟨ sym y≈x ⟩ y≈z
```

A host of similar such combinators can be found within the RATH-Agda library.

5 Properties of Sums and Products

This module is for those domain-ubiquitous properties that, disappointingly, we could not locate in the standard library. —The standard library needs some sort of "table of contents *with* subsection" to make it easier to know of what is available.

This module re-exports (some of) the contents of the standard library's Data.Product and Data.Sum.

```
module DataProperties where
open import Level renaming (suc to lsuc; zero to lzero)
open import Function using (id; _∘_; const)
open import EqualityCombinators
open import Data.Product public using (_×_; proj₁; proj₂; Σ; _,_; swap; uncurry) renaming (map to _×₁_; <_,_> to ⟨_,_⟩)
open import Data.Sum public using (inj₁; inj₂; [_,_]) renaming (map to _⊎₁_)
open import Data.Nat using (ℕ; zero; suc)
```

Precedence Levels

The standard library assigns precedence level of 1 for the infix operator _⊎_, which is rather odd since infix operators ought to have higher precedence that equality combinators, yet the standard library assigns _≈⟨_⟩_ a precedence level of 2. The usage of these two —e.g. in CommMonoid.lagda— causes an annoying number of parentheses and so we reassign the level of the infix operator to avoid such a situation.

```
infixr 3 _⊎_
_⊎_ = Data.Sum._⊎_
```

5.1 Generalised Bot and Top

To avoid a flurry of lift's, and for the sake of clarity, we define level-polymorphic empty and unit types.

```
open import Level
data ⊥ {ℓ : Level} : Set ℓ where
⊥-elim : {a ℓ : Level} {A : Set a} → ⊥ {ℓ} → A
⊥-elim ()
record ⊤ {ℓ : Level} : Set ℓ where
  constructor tt
```

5.2 Sums

Just as _⊎_ takes types to types, its "map" variant $\_\uplus_1\_$ takes functions to functions and is a functorial congruence: It preserves identity, distributes over composition, and preserves extensional equality.

```
⊎-id : {a b : Level} {A : Set a} {B : Set b} → id ⊎₁ id ≐ id {A = A ⊎ B}
⊎-id = [ ≐-refl , ≐-refl ]
⊎-∘ : {a b c a' b' c' : Level}
  {A : Set a} {A' : Set a'} {B : Set b} {B' : Set b'} {C' : Set c} {C : Set c'}
  {f : A → A'} {g : B → B'} {f' : A' → C} {g' : B' → C'}
  → (f' ∘ f) ⊎₁ (g' ∘ g) ≐ (f' ⊎₁ g') ∘ (f ⊎₁ g)   -- aka "the exchange rule for sums"
⊎-∘ = [ ≐-refl , ≐-refl ]
⊎-cong : {a b c d : Level} {A : Set a} {B : Set b} {C : Set c} {D : Set d} {f f' : A → C} {g g' : B → D}
  → f ≐ f' → g ≐ g' → f ⊎₁ g ≐ f' ⊎₁ g'
⊎-cong f≈f' g≈g' = [ ∘-≐-cong₂ inj₁ f≈f' , ∘-≐-cong₂ inj₂ g≈g' ]
```

Composition post-distributes into casing,

```
∘-[,] : {a b c d : Level} {A : Set a} {B : Set b} {C : Set c} {D : Set d} {f : A → C} {g : B → C} {h : C → D}
  → h ∘ [ f , g ] ≐ [ h ∘ f , h ∘ g ]   -- aka "fusion"
∘-[,] = [ ≐-refl , ≐-refl ]
```

It is common that a data-type constructor D : Set → Set allows us to extract elements of the underlying type and so we have a natural transfomation D ⟶ **I**, where **I** is the identity functor. These kind of results will occur for our other simple data-strcutres as well. In particular, this is the case for D A = 2 × A = A ⊎ A:

```
from⊎ : {ℓ : Level} {A : Set ℓ} → A ⊎ A → A
from⊎ = [ id , id ]
  -- from⊎ is a natural transformation
  --
from⊎-nat : {a b : Level} {A : Set a} {B : Set b} {f : A → B} → f ∘ from⊎ ≐ from⊎ ∘ (f ⊎₁ f)
from⊎-nat = [ ≐-refl , ≐-refl ]
  -- from⊎ is injective and so is pre-invertible,
  --
from⊎-preInverse : {a b : Level} {A : Set a} {B : Set b} → id ≐ from⊎ {A = A ⊎ B} ∘ (inj₁ ⊎₁ inj₂)
from⊎-preInverse = [ ≐-refl , ≐-refl ]
```

[**MA: insert:** A brief mention about co-monads?]

5.3 Products

Dual to from⊎, a natural transformation 2 ×_ ⟶ **I**, is diag, the transformation **I** ⟶ $\_^2$.

```
diag : {ℓ : Level} {A : Set ℓ} (a : A) → A × A
diag a = a , a
```

[**MA: insert:** A brief mention of Haskell's const, which is diag curried. Also something about K combinator?]

```
open import Data.Nat.Properties
suc-inj : ∀ {i j} → ℕ.suc i ≡ ℕ.suc j → i ≡ j
suc-inj = cancel-+-left (ℕ.suc ℕ.zero)
```

or

```
suc-inj {0} _≡_.refl = _≡_.refl
suc-inj {ℕ.suc i} _≡_.refl = _≡_.refl
```

6 Two Sorted Structures

So far we have been considering algebraic structures with only one underlying carrier set, however programmers are faced with a variety of different types at the same time, and the graph structure between them, and so we consider briefly consider two sorted structures by starting the simplest possible case: Two type and no required interaction whatsoever between them.

```
module Structures.TwoSorted where
open import Level renaming (suc to lsuc; zero to lzero)
open import Categories.Category    using (Category)
open import Categories.Functor     using (Functor)
open import Categories.Adjunction using (Adjunction)
open import Categories.Agda        using (Sets)
open import Function               using (id; _∘_; const)
open import Function2              using (_$ᵢ)
open import Forget
open import EqualityCombinators
open import DataProperties
```

6.1 Definitions

A TwoSorted type is just a pair of sets in the same universe —in the future, we may consider those in different levels.

```
record TwoSorted ℓ : Set (lsuc ℓ) where
  constructor MkTwo
  field
    One : Set ℓ
    Two : Set ℓ
open TwoSorted
```

Unastionishngly, a morphism between such types is a pair of functions between the *multiple* underlying carriers.

```
record Hom {ℓ} (Src Tgt : TwoSorted ℓ) : Set ℓ where
  constructor MkHom
  field
    one : One Src → One Tgt
    two : Two Src → Two Tgt
open Hom
```

6.2 Category and Forgetful Functors

We are using pairs of object and pairs of morphisms which are known to form a category:

```
Twos : (ℓ : Level) → Category (lsuc ℓ) ℓ ℓ
Twos ℓ = record
```

```
{Obj       = TwoSorted ℓ
; _⇒_      = Hom
; _≡_      = λ F G → one F ≐ one G × two F ≐ two G
; id       = MkHom id id
; _∘_      = λ F G → MkHom (one F ∘ one G) (two F ∘ two G)
; assoc    = ≐-refl , ≐-refl
; identityˡ = ≐-refl , ≐-refl
; identityʳ = ≐-refl , ≐-refl
; equiv    = record
   {refl  = ≐-refl , ≐-refl
   ; sym  = λ {(oneEq , twoEq) → ≐-sym oneEq , ≐-sym twoEq}
   ; trans = λ {(oneEq₁ , twoEq₁) (oneEq₂ , twoEq₂) → ≐-trans oneEq₁ oneEq₂ , ≐-trans twoEq₁ twoEq₂}
   }
; ∘-resp-≡ = λ {(g≈₁k , g≈₂k) (f≈₁h , f≈₂h) → ∘-resp-≐ g≈₁k f≈₁h , ∘-resp-≐ g≈₂k f≈₂h}
}
```

The naming Twos is to be consistent with the category theory library we are using, which names the category of sets and functions by Sets.

We can forget about the first sort or the second to arrive at our starting category and so we have two forgetful functors.

```
Forget : (ℓ : Level) → Functor (Twos ℓ) (Sets ℓ)
Forget ℓ = record
  {F₀              = TwoSorted.One
  ; F₁             = Hom.one
  ; identity       = ≡.refl
  ; homomorphism   = ≡.refl
  ; F-resp-≡ = λ {(F≈₁G , F≈₂G) {x} → F≈₁G x}
  }
Forget² : (ℓ : Level) → Functor (Twos ℓ) (Sets ℓ)
Forget² ℓ = record
  {F₀              = TwoSorted.Two
  ; F₁             = Hom.two
  ; identity       = ≡.refl
  ; homomorphism   = ≡.refl
  ; F-resp-≡ = λ {(F≈₁G , F≈₂G) {x} → F≈₂G x}
  }
```

6.3 Free and CoFree

Given a type, we can pair it with the empty type or the singelton type and so we have a free and a co-free constructions. Intuitively, the first is free since the singelton type is the smallest type we can adjoin to obtain a Twos object, whereas ⊤ is the "largest" type we adjoin to obtain a Twos object. This is one way that the unit and empty types naturaly arise.

```
Free : (ℓ : Level) → Functor (Sets ℓ) (Twos ℓ)
Free ℓ = record
  {F₀              = λ A → MkTwo A ⊥
  ; F₁             = λ f → MkHom f id
  ; identity       = ≐-refl , ≐-refl
  ; homomorphism   = ≐-refl , ≐-refl
  ; F-resp-≡ = λ f≈g → (λ x → f≈g {x}) , ≐-refl
  }
```

```
Cofree : (ℓ : Level) → Functor (Sets ℓ) (Twos ℓ)
Cofree ℓ = record
  {F₀            = λ A → MkTwo A ⊤
  ; F₁           = λ f → MkHom f id
  ; identity     = ≐-refl , ≐-refl
  ; homomorphism = ≐-refl , ≐-refl
  ; F-resp-≡ = λ f≈g → (λ x → f≈g {x}) , ≐-refl
  }
  -- Dually, ( also shorter due to eta reduction )
Free² : (ℓ : Level) → Functor (Sets ℓ) (Twos ℓ)
Free² ℓ = record
  {F₀            = MkTwo ⊥
  ; F₁           = MkHom id
  ; identity     = ≐-refl , ≐-refl
  ; homomorphism = ≐-refl , ≐-refl
  ; F-resp-≡ = λ f≈g → ≐-refl , λ x → f≈g {x}
  }
Cofree² : (ℓ : Level) → Functor (Sets ℓ) (Twos ℓ)
Cofree² ℓ = record
  {F₀            = MkTwo ⊤
  ; F₁           = MkHom id
  ; identity     = ≐-refl , ≐-refl
  ; homomorphism = ≐-refl , ≐-refl
  ; F-resp-≡ = λ f≈g → ≐-refl , λ x → f≈g {x}
  }
```

6.4 Adjunction Proofs

Now for the actual proofs that the Free and Cofree functors are deserving of their names.

```
Left : (ℓ : Level) → Adjunction (Free ℓ) (Forget ℓ)
Left ℓ = record
  {unit = record
    {η = λ _ → id
    ; commute = λ _ → ≡.refl
    }
  ; counit = record
    {η = λ _ → MkHom id (λ {()})
    ; commute = λ f → ≐-refl , (λ {()})
    }
  ; zig = ≐-refl , (λ {()})
  ; zag = ≡.refl
  }
Right : (ℓ : Level) → Adjunction (Forget ℓ) (Cofree ℓ)
Right ℓ = record
  {unit = record
    {η = λ _ → MkHom id (λ _ → tt)
    ; commute = λ _ → ≐-refl , ≐-refl
    }
  ; counit = record {η = λ _ → id; commute = λ _ → ≡.refl}
  ; zig    = ≡.refl
  ; zag    = ≐-refl , λ {tt → ≡.refl}
  }
  -- Dually,
```

```
Left² : (ℓ : Level) → Adjunction (Free² ℓ) (Forget² ℓ)
Left² ℓ = record
  {unit = record
    {η = λ _ → id
    ;commute = λ _ → ≡.refl
    }
  ;counit = record
    {η = λ _ → MkHom (λ {()}) id
    ;commute = λ f → (λ {()}) , ≐-refl
    }
  ;zig = (λ {()}) , ≐-refl
  ;zag = ≡.refl
  }
Right² : (ℓ : Level) → Adjunction (Forget² ℓ) (Cofree² ℓ)
Right² ℓ = record
  {unit = record
    {η = λ _ → MkHom (λ _ → tt) id
    ;commute = λ _ → ≐-refl , ≐-refl
    }
  ;counit = record {η = λ _ → id; commute = λ _ → ≡.refl}
  ;zig    = ≡.refl
  ;zag    = (λ {tt → ≡.refl}) , ≐-refl
  }
```

6.5 Merging is adjoint to duplication

The category of sets contains products and so TwoSorted algebras can be represented there and, moreover, this is adjoint to duplicating a type to obtain a TwoSorted algebra.

```
  -- The category of Sets has products and so the TwoSorted type can be reified there.
Merge : (ℓ : Level) → Functor (Twos ℓ) (Sets ℓ)
Merge ℓ = record
  {F₀              = λ S → One S × Two S
  ;F₁              = λ F → one F ×₁ two F
  ;identity        = ≡.refl
  ;homomorphism    = ≡.refl
  ;F-resp-≡ = λ {(F≈₁G , F≈₂G) {x , y} → ≡.cong₂ _,_ (F≈₁G x) (F≈₂G y)}
  }
  -- Every set gives rise to its square as a TwoSorted type.
Dup : (ℓ : Level) → Functor (Sets ℓ) (Twos ℓ)
Dup ℓ = record
  {F₀              = λ A → MkTwo A A
  ;F₁              = λ f → MkHom f f
  ;identity        = ≐-refl , ≐-refl
  ;homomorphism    = ≐-refl , ≐-refl
  ;F-resp-≡ = λ F≈G → diag (λ _ → F≈G)
  }
```

Then the proof that these two form the desired adjunction

```
Right₂ : (ℓ : Level) → Adjunction (Dup ℓ) (Merge ℓ)
Right₂ ℓ = record
  {unit   = record {η = λ _ → diag; commute = λ _ → ≡.refl}
  ;counit = record {η = λ _ → MkHom proj₁ proj₂; commute = λ _ → ≐-refl , ≐-refl}
```

```
  ; zig      = ≐-refl , ≐-refl
  ; zag      = ≡.refl
  }
```

6.6 Duplication also has a left adjoint

The category of sets admits sums and so an alternative is to represet a TwoSorted algebra as a sum, and moreover this is adjoint to the aforementioned duplication functor.

```
Choice : (ℓ : Level) → Functor (Twos ℓ) (Sets ℓ)
Choice ℓ = record
  {F₀              = λ S → One S ⊎ Two S
  ; F₁             = λ F → one F ⊎₁ two F
  ; identity       = ⊎-id $ᵢ
  ; homomorphism   = λ {{x = x} → ⊎-∘ x}
  ; F-resp-≡  =  λ F≈G {x} → uncurry ⊎-cong F≈G x
  }
Left₂ : (ℓ : Level) → Adjunction (Choice ℓ) (Dup ℓ)
Left₂ ℓ = record
  {unit      = record {η = λ _ → MkHom inj₁ inj₂; commute = λ _ → ≐-refl , ≐-refl}
  ; counit   = record {η = λ _ → from⊎; commute = λ _ {x} → (≡.sym ∘ from⊎-nat) x}
  ; zig      = λ {{_} {x} → from⊎-preInverse x}
  ; zag      = ≐-refl , ≐-refl
  }
```

7 Binary Heterogeneous Relations — [MA: *What* named *data structure do these correspond to in programming?*]

We consider two sorted algebras endowed with a binary heterogeneous relation. An example of such a structure is a graph, or network, which has a sort for edges and a sort for nodes and an incidence relation.

```
module Structures.Rel where
open import Level renaming (suc to lsuc; zero to lzero; _⊔_ to _⊍_)
open import Categories.Category    using (Category)
open import Categories.Functor     using (Functor)
open import Categories.Adjunction using (Adjunction)
open import Categories.Agda        using (Sets)
open import Function               using (id; _∘_; const)
open import Function2              using (_$ᵢ)
open import Forget
open import EqualityCombinators
open import DataProperties
open import Structures.TwoSorted using (TwoSorted; Twos; MkTwo) renaming (Hom to TwoHom; MkHom to MkTwoHom)
```

7.1 Definitions

We define the structure involved, along with a notational convenience:

```
record HetroRel ℓ ℓ′ : Set (lsuc (ℓ ⊍ ℓ′)) where
  constructor MkHRel
```

```
  field
    One : Set ℓ
    Two : Set ℓ
    Rel : One → Two → Set ℓ′
open HetroRel
relOp = HetroRel.Rel
syntax relOp A x y = x ⟨ A ⟩ y
```

Then define the strcture-preserving operations,

```
record Hom {ℓ ℓ′} (Src Tgt : HetroRel ℓ ℓ′) : Set (ℓ ⊍ ℓ′) where
  constructor MkHom
  field
    one : One Src → One Tgt
    two : Two Src → Two Tgt
    shift : {x : One Src} {y : Two Src} → x ⟨ Src ⟩ y → one x ⟨ Tgt ⟩ two y
open Hom
```

7.2 Category and Forgetful Functors

That these structures form a two-sorted algebraic category can easily be witnessed.

```
Rels : (ℓ ℓ′ : Level) → Category (lsuc (ℓ ⊍ ℓ′)) (ℓ ⊍ ℓ′) ℓ
Rels ℓ ℓ′ = record
  {Obj       = HetroRel ℓ ℓ′
  ; _⇒_      = Hom
  ; _≡_      = λ F G → one F ≐ one G × two F ≐ two G
  ; id       = MkHom id id id
  ; _∘_      = λ F G → MkHom (one F ∘ one G) (two F ∘ two G) (shift F ∘ shift G)
  ; assoc    = ≐-refl , ≐-refl
  ; identityˡ = ≐-refl , ≐-refl
  ; identityʳ = ≐-refl , ≐-refl
  ; equiv    = record
    {refl  = ≐-refl , ≐-refl
    ; sym   = λ {(oneEq , twoEq) → ≐-sym oneEq , ≐-sym twoEq}
    ; trans = λ {(oneEq₁ , twoEq₁) (oneEq₂ , twoEq₂) → ≐-trans oneEq₁ oneEq₂ , ≐-trans twoEq₁ twoEq₂}
    }
  ; ∘-resp-≡ = λ {(g≈₁k , g≈₂k) (f≈₁h , f≈₂h) → ∘-resp-≐ g≈₁k f≈₁h , ∘-resp-≐ g≈₂k f≈₂h}
  }
```

We can forget about the first sort or the second to arrive at our starting category and so we have two forgetful functors. Moreover, we can simply forget about the relation to arrive at the two-sorted category :-)

```
Forget¹ : (ℓ ℓ′ : Level) → Functor (Rels ℓ ℓ′) (Sets ℓ)
Forget¹ ℓ ℓ′ = record
  {F₀            = HetroRel.One
  ; F₁           = Hom.one
  ; identity     = ≡.refl
  ; homomorphism = ≡.refl
  ; F-resp-≡ = λ {(F≈₁G , F≈₂G) {x} → F≈₁G x}
  }
Forget² : (ℓ ℓ′ : Level) → Functor (Rels ℓ ℓ′) (Sets ℓ)
Forget² ℓ ℓ′ = record
  {F₀            = HetroRel.Two
```

```
  ; F₁              = Hom.two
  ; identity        = ≡.refl
  ; homomorphism    = ≡.refl
  ; F-resp-≡  =  λ {(F≈₁G , F≈₂G) {x} → F≈₂G x}
  }
  -- Whence, Rels is a subcategory of Twos
Forget³ : (ℓ ℓ′ : Level) → Functor (Rels ℓ ℓ′) (Twos ℓ)
Forget³ ℓ ℓ′ = record
  {F₀               = λ S → MkTwo (One S) (Two S)
  ; F₁              = λ F → MkTwoHom (one F) (two F)
  ; identity        = ≐-refl , ≐-refl
  ; homomorphism    = ≐-refl , ≐-refl
  ; F-resp-≡  =  id
  }
```

7.3 Free and CoFree Functors

Given a (two)type, we can pair it with the empty type or the singleton type and so we have a free and a co-free constructions. Intuitively, the empty type denotes the empty relation which is the smallest relation and so a free construction; whereas, the singleton type denotes the “always true” relation which is the largest binary relation and so a cofree construction.

Candidate adjoints to forgetting the *first* component of a Rels

```
Free¹ : (ℓ ℓ′ : Level) → Functor (Sets ℓ) (Rels ℓ ℓ′)
Free¹ ℓ ℓ′ = record
  {F₀               = λ A → MkHRel A ⊥ (λ {_ ()})
  ; F₁              = λ f → MkHom f id (λ {{y = ()}})
  ; identity        = ≐-refl , ≐-refl
  ; homomorphism    = ≐-refl , ≐-refl
  ; F-resp-≡  =  λ f≈g → (λ x → f≈g {x}) , ≐-refl
  }
  -- (MkRel X ⊥ ⊥ ⟶ Alg) ≅ (X ⟶ One Alg)
Left¹ : (ℓ ℓ′ : Level) → Adjunction (Free¹ ℓ ℓ′) (Forget¹ ℓ ℓ′)
Left¹ ℓ ℓ′ = record
  {unit    = record
    {η  =  λ _ → id
    ; commute  =  λ _ → ≡.refl
    }
  ; counit  =  record
    {η  =  λ A → MkHom (λ z → z) (λ {()}) (λ {x} {})
    ; commute  =  λ f → ≐-refl , (λ ())
    }
  ; zig  =  ≐-refl , (λ ())
  ; zag  =  ≡.refl
  }
CoFree¹ : (ℓ : Level) → Functor (Sets ℓ) (Rels ℓ ℓ)
CoFree¹ ℓ = record
  {F₀               = λ A → MkHRel A ⊤ (λ _ _ → A)
  ; F₁              = λ f → MkHom f id f
  ; identity        = ≐-refl , ≐-refl
  ; homomorphism    = ≐-refl , ≐-refl
```

```
  ; F-resp-≡  =  λ f≈g → (λ x → f≈g {x}) , ≐-refl
  }
  -- (One Alg ⟶ X) ≅ (Alg ⟶ MkRel X ⊤ (λ _ _ → X)
Right¹ : (ℓ : Level) → Adjunction (Forget¹ ℓ ℓ) (CoFree¹ ℓ)
Right¹ ℓ  =  record
  {unit  =  record
    {η  =  λ _ → MkHom id (λ _ → tt) (λ {x} {y} _ → x)
    ; commute  =  λ _ → ≐-refl , (λ x → ≡.refl)
    }
  ; counit  =  record {η  =  λ _ → id; commute  =  λ _ → ≡.refl}
  ; zig     =  ≡.refl
  ; zag     =  ≐-refl , λ {tt → ≡.refl}
  }
  -- Another cofree functor:
CoFree¹′ : (ℓ : Level) → Functor (Sets ℓ) (Rels ℓ ℓ)
CoFree¹′ ℓ  =  record
  {F₀              =  λ A → MkHRel A ⊤ (λ _ _ → ⊤)
  ; F₁             =  λ f → MkHom f id id
  ; identity       =  ≐-refl , ≐-refl
  ; homomorphism   =  ≐-refl , ≐-refl
  ; F-resp-≡  =  λ f≈g → (λ x → f≈g {x}) , ≐-refl
  }
  -- (One Alg ⟶ X) ≅ (Alg ⟶ MkRel X ⊤ (λ _ _ → ⊤)
Right¹′ : (ℓ : Level) → Adjunction (Forget¹ ℓ ℓ) (CoFree¹′ ℓ)
Right¹′ ℓ  =  record
  {unit  =  record
    {η  =  λ _ → MkHom id (λ _ → tt) (λ {x} {y} _ → tt)
    ; commute  =  λ _ → ≐-refl , (λ x → ≡.refl)
    }
  ; counit  =  record {η  =  λ _ → id; commute  =  λ _ → ≡.refl}
  ; zig     =  ≡.refl
  ; zag     =  ≐-refl , λ {tt → ≡.refl}
  }
```

But wait, adjoints are necessarily unique, up to isomorphism, whence $\mathsf{CoFree}^1 \cong \mathsf{Cofree}^{1\prime}$. Intuitively, the relation part is a "subset" of the given carriers and when one of the carriers is a singleton then the largest relation is the universal relation which can be seen as either the first non-singleton carrier or the "always-true" relation which happens to be formalized by ignoring its arguments and going to a singleton set.

Candidate adjoints to forgetting the *second* component of a Rels

```
Free² : (ℓ : Level) → Functor (Sets ℓ) (Rels ℓ ℓ)
Free² ℓ  =  record
  {F₀              =    λ A → MkHRel ⊥ A (λ ())
  ; F₁             =    λ f → MkHom id f (λ {})
  ; identity       =    ≐-refl , ≐-refl
  ; homomorphism   =    ≐-refl , ≐-refl
  ; F-resp-≡  =  λ F≈G → ≐-refl , (λ x → F≈G {x})
  }
  -- (MkRel ⊥ X ⊥ ⟶ Alg) ≅ (X ⟶ Two Alg)
Left² : (ℓ : Level) → Adjunction (Free² ℓ) (Forget² ℓ ℓ)
Left² ℓ  =  record
  {unit  =  record
```

```
      {η = λ _ → id
      ; commute = λ _ → ≡.refl
      }
    ; counit = record
      {η = λ _ → MkHom (λ ()) id (λ {})
      ; commute = λ f → (λ ()) , ≐-refl
      }
    ; zig = (λ ()) , ≐-refl
    ; zag = ≡.refl
    }
CoFree² : (ℓ : Level) → Functor (Sets ℓ) (Rels ℓ ℓ)
CoFree² ℓ = record
    {F₀            =   λ A → MkHRel ⊤ A (λ _ _ → ⊤)
    ; F₁           =   λ f → MkHom id f id
    ; identity     =   ≐-refl , ≐-refl
    ; homomorphism =   ≐-refl , ≐-refl
    ; F-resp-≡ = λ F≈G → ≐-refl , (λ x → F≈G {x})
    }
  -- (Two Alg ⟶ X) ≅ (Alg ⟶ ⊤ X ⊤
Right² : (ℓ : Level) → Adjunction (Forget² ℓ ℓ) (CoFree² ℓ)
Right² ℓ = record
    {unit = record
      {η = λ _ → MkHom (λ _ → tt) id (λ _ → tt)
      ; commute = λ f → ≐-refl , ≐-refl
      }
    ; counit = record
      {η = λ _ → id
      ; commute = λ _ → ≡.refl
      }
    ; zig = ≡.refl
    ; zag = (λ {tt → ≡.refl}) , ≐-refl
    }
```

Candidate adjoints to forgetting the *third* component of a Rels

```
Free³ : (ℓ : Level) → Functor (Twos ℓ) (Rels ℓ ℓ)
Free³ ℓ = record
    {F₀            =   λ S → MkHRel (One S) (Two S) (λ _ _ → ⊥)
    ; F₁           =   λ f → MkHom (one f) (two f) id
    ; identity     =   ≐-refl , ≐-refl
    ; homomorphism =   ≐-refl , ≐-refl
    ; F-resp-≡ = id
    } where open TwoSorted; open TwoHom
  -- (MkTwo X Y → Alg without Rel) ≅ (MkRel X Y ⊥ ⟶ Alg)
Left³ : (ℓ : Level) → Adjunction (Free³ ℓ) (Forget³ ℓ ℓ)
Left³ ℓ = record
    {unit = record
      {η = λ A → MkTwoHom id id
      ; commute = λ F → ≐-refl , ≐-refl
      }
    ; counit = record
      {η = λ A → MkHom id id (λ ())
      ; commute = λ F → ≐-refl , ≐-refl
      }
```

```
  ; zig  =  ≐-refl , ≐-refl
  ; zag  =  ≐-refl , ≐-refl
  }


CoFree³ : (ℓ : Level) → Functor (Twos ℓ) (Rels ℓ ℓ)
CoFree³ ℓ = record
  {F₀              =   λ S → MkHRel (One S) (Two S) (λ _ _ → ⊤)
  ; F₁             =   λ f → MkHom (one f) (two f) id
  ; identity       =   ≐-refl , ≐-refl
  ; homomorphism   =   ≐-refl , ≐-refl
  ; F-resp-≡  =  id
  } where open TwoSorted; open TwoHom

  -- (Alg without Rel ⟶ MkTwo X Y) ≅ (Alg ⟶ MkRel X Y ⊤)
Right³ : (ℓ : Level) → Adjunction (Forget³ ℓ ℓ) (CoFree³ ℓ)
Right³ ℓ = record
  {unit = record
    {η = λ A → MkHom id id (λ _ → tt)
    ; commute = λ F → ≐-refl , ≐-refl
    }
  ; counit = record
    {η = λ A → MkTwoHom id id
    ; commute = λ F → ≐-refl , ≐-refl
    }
  ; zig = ≐-refl , ≐-refl
  ; zag = ≐-refl , ≐-refl
  }
CoFree³′ : (ℓ : Level) → Functor (Twos ℓ) (Rels ℓ ℓ)
CoFree³′ ℓ = record
  {F₀              =   λ S → MkHRel (One S) (Two S) (λ _ _ → One S × Two S)
  ; F₁             =   λ F → MkHom (one F) (two F) (one F ×₁ two F)
  ; identity       =   ≐-refl , ≐-refl
  ; homomorphism   =   ≐-refl , ≐-refl
  ; F-resp-≡  =  id
  } where open TwoSorted; open TwoHom

  -- (Alg without Rel ⟶ MkTwo X Y) ≅ (Alg ⟶ MkRel X Y X×Y)
Right³′ : (ℓ : Level) → Adjunction (Forget³ ℓ ℓ) (CoFree³′ ℓ)
Right³′ ℓ = record
  {unit = record
    {η = λ A → MkHom id id (λ {x} {y} x˜y → x , y)
    ; commute = λ F → ≐-refl , ≐-refl
    }
  ; counit = record
    {η = λ A → MkTwoHom id id
    ; commute = λ F → ≐-refl , ≐-refl
    }
  ; zig = ≐-refl , ≐-refl
  ; zag = ≐-refl , ≐-refl
  }
```

But wait, adjoints are necessarily unique, up to isomorphism, whence $\mathsf{CoFree}^3 \cong \mathsf{CoFree}^{3\prime}$. Intuitively, the relation part is a "subset" of the given carriers and so the largest relation is the universal relation which can be seen as the product of the carriers or the "always-true" relation which happens to be formalized by ignoring its arguments and going to a singleton set.

7.4 ???

It remains to port over results such as Merge, Dup, and Choice from Twos to Rels.

Also to consider: sets with an equivalence relation; whence propositional equality.

The category of sets contains products and so TwoSorted algebras can be represented there and, moreover, this is adjoint to duplicating a type to obtain a TwoSorted algebra.

```
  -- The category of Sets has products and so the TwoSorted type can be reified there.
Merge : (ℓ : Level) → Functor (Twos ℓ) (Sets ℓ)
Merge ℓ = record
  {F₀             = λ S → One S × Two S
  ;F₁             = λ F → one F ×₁ two F
  ;identity       = ≡.refl
  ;homomorphism   = ≡.refl
  ;F-resp-≡  = λ {(F≈₁G , F≈₂G) {x , y} → ≡.cong₂ _,_ (F≈₁G x) (F≈₂G y)}
  }
  -- Every set gives rise to its square as a TwoSorted type.
Dup : (ℓ : Level) → Functor (Sets ℓ) (Twos ℓ)
Dup ℓ = record
  {F₀             = λ A → MkTwo A A
  ;F₁             = λ f → MkHom f f
  ;identity       = ≐-refl , ≐-refl
  ;homomorphism   = ≐-refl , ≐-refl
  ;F-resp-≡  = λ F≈G → diag (λ _ → F≈G)
  }
```

Then the proof that these two form the desired adjunction

```
Right₂ : (ℓ : Level) → Adjunction (Dup ℓ) (Merge ℓ)
Right₂ ℓ = record
  {unit    = record {η = λ _ → diag; commute = λ _ → ≡.refl}
  ;counit  = record {η = λ _ → MkHom proj₁ proj₂; commute = λ _ → ≐-refl , ≐-refl}
  ;zig     = ≐-refl , ≐-refl
  ;zag     = ≡.refl
  }
```

The category of sets admits sums and so an alternative is to represet a TwoSorted algebra as a sum, and moreover this is adjoint to the aforementioned duplication functor.

```
Choice : (ℓ : Level) → Functor (Twos ℓ) (Sets ℓ)
Choice ℓ = record
  {F₀             = λ S → One S ⊎ Two S
  ;F₁             = λ F → one F ⊎₁ two F
  ;identity       = ⊎-id $ᵢ
  ;homomorphism   = λ {{x = x} → ⊎-∘ x}
  ;F-resp-≡  = λ F≈G {x} → uncurry ⊎-cong F≈G x
  }
Left₂ : (ℓ : Level) → Adjunction (Choice ℓ) (Dup ℓ)
Left₂ ℓ = record
  {unit     = record {η = λ _ → MkHom inj₁ inj₂; commute = λ _ → ≐-refl , ≐-refl}
  ;counit   = record {η = λ _ → from⊎; commute = λ _ {x} → (≡.sym ∘ from⊎-nat) x}
  ;zig      = λ {{_} {x} → from⊎-preInverse x}
  ;zag      = ≐-refl , ≐-refl
  }
```

8 Pointed Algebras: Nullable Types

We consider the theory of *pointed algebras* which consist of a type along with an elected value of that type.[1] Software engineers encounter such scenarios all the time in the case of an object-type and a default value of a "null", or undefined, object. In the more explicit setting of pure functional programming, this concept arises in the form of Maybe, or Option types.

Some programming languages, such as C# for example, provide a default keyword to access a default value of a given data type.

[**MA:** **insert:** Haskell's typeclass analogue of default?]

[**MA:** *Perhaps discuss "types as values" and the subtle issue of how pointed algebras are completely different than classes in an imperative setting.*]

```
module Structures.Pointed where

open import Level renaming (suc to lsuc; zero to lzero)
open import Categories.Category using (Category; module Category)
open import Categories.Functor using (Functor)
open import Categories.Adjunction using (Adjunction)
open import Categories.NaturalTransformation using (NaturalTransformation)
open import Categories.Agda using (Sets)
open import Function using (id; _∘_)
open import Data.Maybe using (Maybe; just; nothing; maybe; maybe′)
open import Forget
open import Data.Empty
open import Relation.Nullary
open import EqualityCombinators
```

8.1 Definition

As mentioned before, a Pointed algebra is a type, which we will refer to by Carrier, along with a value, or point, of that type.

```
record Pointed {a} : Set (lsuc a) where
  constructor MkPointed
  field
    Carrier : Set a
    point   : Carrier
open Pointed
```

Unsurprisingly, a "structure preserving operation" on such structures is a function between the underlying carriers that takes the source's point to the target's point.

```
record Hom {ℓ} (X Y : Pointed {ℓ}) : Set ℓ where
  constructor MkHom
  field
    mor          : Carrier X → Carrier Y
    preservation : mor (point X) ≡ point Y
open Hom
```

[1] Note that this definition is phrased as a "dependent product"!

8.2 Category and Forgetful Functors

Since there is only one type, or sort, involved in the definition, we may hazard these structures as "one sorted algebras":

```
oneSortedAlg : ∀ {ℓ} → OneSortedAlg ℓ
oneSortedAlg = record
  {Alg         = Pointed
  ; Carrier    = Carrier
  ; Hom        = Hom
  ; mor        = mor
  ; comp       = λ F G → MkHom (mor F ∘ mor G) (≡.cong (mor F) (preservation G) ⟨≡≡⟩ preservation F)
  ; comp-is-∘  = ≐-refl
  ; Id         = MkHom id ≡.refl
  ; Id-is-id   = ≐-refl
  }
```

From which we immediately obtain a category and a forgetful functor.

```
Pointeds : (ℓ : Level) → Category (lsuc ℓ) ℓ ℓ
Pointeds ℓ = oneSortedCategory ℓ oneSortedAlg
Forget : (ℓ : Level) → Functor (Pointeds ℓ) (Sets ℓ)
Forget ℓ = mkForgetful ℓ oneSortedAlg
```

The naming Pointeds is to be consistent with the category theory library we are using, which names the category of sets and functions by Sets. That is, the category name is the objects' name suffixed with an 's'.

Of-course, as hinted in the introduction, this structure —as are many— is defined in a dependent fashion and so we have another forgetful functor:

```
open import Data.Product
Forgetᴰ : (ℓ : Level) → Functor (Pointeds ℓ) (Sets ℓ)
Forgetᴰ ℓ = record {F₀ = λ P → Σ (Carrier P) (λ x → x ≡ point P)
  ; F₁ = λ {P} {Q} F → λ {(val , val≡ptP) → mor F val , (≡.cong (mor F) val≡ptP ⟨≡≡⟩ preservation F)}
  ; identity = λ {P} → λ {{val , val≡ptP} → ≡.cong (λ x → val , x) (≡.proof-irrelevance _ _)}
  ; homomorphism = λ {P} {Q} {R} {F} {G} → λ {{val , val≡ptP} → ≡.cong (λ x → mor G (mor F val) , x) (≡.proof-irrelevance _ _)}
  ; F-resp-≡ = λ {P} {Q} {F} {G} F≈G → λ {{val , val≡ptP} → {!≡.cong₂ _,_ (F≈G val) ?!}}
  }
```

That is, we "only remember the point".

[**MA: insert:** An adjoint to this functor?]

8.3 A Free Construction

As discussed earlier, the prime example of pointed algebras are the optional types, and this claim can be realised as a functor:

```
Free : (ℓ : Level) → Functor (Sets ℓ) (Pointeds ℓ)
Free ℓ = record
  {F₀              = λ A → MkPointed (Maybe A) nothing
  ; F₁             = λ f → MkHom (maybe (just ∘ f) nothing) ≡.refl
  ; identity       = maybe ≐-refl ≡.refl
  ; homomorphism   = maybe ≐-refl ≡.refl
  ; F-resp-≡ = λ F≡G → maybe (∘-resp-≐ (≐-refl {x = just}) (λ x → F≡G {x})) ≡.refl
  }
```

Which is indeed deserving of its name:

```
MaybeLeft : (ℓ : Level) → Adjunction (Free ℓ) (Forget ℓ)
MaybeLeft ℓ = record
  {unit        = record {η = λ _ → just; commute = λ _ → ≡.refl}
  ; counit     = record
    {η         = λ X → MkHom (maybe id (point X)) ≡.refl
    ; commute  = maybe ≐-refl ∘ ≡.sym ∘ preservation
    }
  ; zig        = maybe ≐-refl ≡.refl
  ; zag        = ≡.refl
  }
```

[**MA:**] *Develop Maybe explicitly so we can "see" how the utility maybe "pops up naturally".*]

While there is a "least" pointed object for any given set, there is, in-general, no "largest" pointed object corresponding to any given set. That is, there is no co-free functor.

```
NoRight : {ℓ : Level} → (CoFree : Functor (Sets ℓ) (Pointeds ℓ)) → ¬ (Adjunction (Forget ℓ) CoFree)
NoRight (record {F₀ = f}) Adjunct = lower (η (counit Adjunct) (Lift ⊥) (point (f (Lift ⊥))))
  where open Adjunction
    open NaturalTransformation
```

9 SetoidSetoid

```
module SetoidSetoid where
open import Level renaming (zero to lzero; suc to lsuc; _⊔_ to _⊍_) hiding (lift)
open import Relation.Binary using (Setoid)
open import DataProperties using (⊤; tt)
open import SetoidEquiv
```

Setoid of setoids SSetoid, and "setoid" of equality proofs. This is an hSet (by fiat), so it is contractible, in that all proofs are the same. [**WK:**] *Where is that fiat in the code? Not distinguishing different isomorphisms is a recipe for disaster.*]

```
_≈S_ : ∀ {a ℓa} {A : Setoid a ℓa} → (e₁ e₂ : Setoid.Carrier A) → Setoid ℓa ℓa
_≈S_ {A = A} e₁ e₂ = let open Setoid A renaming (_≈_ to _≈ₛ_) in
  record {Carrier = e₁ ≈ₛ e₂; _≈_ = λ _ _ → ⊤
    ; isEquivalence = record {refl = tt; sym = λ _ → tt; trans = λ _ _ → tt}}
SSetoid : (ℓ o : Level) → Setoid (lsuc o ⊍ lsuc ℓ) (o ⊍ ℓ)
SSetoid ℓ o = record
  {Carrier = Setoid ℓ o
  ; _≈_ = _≅_
  ; isEquivalence = record {refl = ≅-refl; sym = ≅-sym; trans = ≅-trans}}
```

10 Some

```
module Some where
open import Level renaming (zero to lzero; suc to lsuc) hiding (lift)
```

```
open import Relation.Binary using (Setoid)
open import Function.Equality using (Π; _⟶_; id; _∘_; _⟨$⟩_)
open import Function using (_$_) renaming (id to id₀; _∘_ to _⊚_)
open import Data.List    using (List; []; _++_; _::_; map)
open import Data.Product using (∃)
open import Data.Nat     using (ℕ; zero; suc)
open import EqualityCombinators
open import DataProperties
open import SetoidEquiv
open import TypeEquiv using (swap₊)
open import SetoidSetoid
open import Relation.Binary.Sum
open import Relation.Binary.PropositionalEquality using (inspect; [_])
```

Setoid based variant of Any.

Quite a bit of this is directly inspired by Data.List.Any and Data.List.Any.Properties.

```
module _ {a ℓa} {A : Setoid a ℓa} (P : A ⟶ SSetoid ℓa ℓa) where
  open Setoid A
  private P₀ = λ e → Setoid.Carrier (Π._⟨$⟩_ P e)
  data Some₀ : List Carrier → Set (a ⊔ ℓa) where
    here : {x : Carrier} {xs : List Carrier} (px : P₀ x) → Some₀ (x :: xs)
    there : {x : Carrier} {xs : List Carrier} (pxs : Some₀ xs) → Some₀ (x :: xs)
    -- inhabitants of Some₀ really are just locations...
    -- could go to Fin (length xs) too.
  toℕ : ∀ {xs} → Some₀ xs → ℕ
  toℕ (here _) = 0
  toℕ (there s) = suc (toℕ s)
    -- proof irrelevance built-in here. We only care that these are the same as members of ℕ
  _~S_ : ∀ {xs} → Some₀ xs → Some₀ xs → Set
  s₁ ~S s₂ = toℕ s₁ ≡ toℕ s₂
  Some : List Carrier → Setoid (ℓa ⊔ a) lzero
  Some xs = record
    {Carrier       = Some₀ xs
    ; _≈_          = _~S_
    ; isEquivalence = record {refl = ≡.refl; sym = ≡.sym; trans = ≡.trans}
    }
≡→Some : {a ℓa : Level} {A : Setoid a ℓa} {P : A ⟶ SSetoid ℓa ℓa}
  {xs ys : List (Setoid.Carrier A)} → xs ≡ ys → Some P xs ≅ Some P ys
≡→Some {A = A} ≡.refl = ≅-refl

module Membership {a ℓ} (S : Setoid a ℓ) where
  open Setoid S renaming (trans to _⟨≈≈⟩_)
  infix 4 _∈₀_ _∈_
  setoid≈ : Carrier → S ⟶ SSetoid ℓ ℓ
  setoid≈ x = record
    {_⟨$⟩_ = λ y → _≈S_ {A = S} x y   -- This is an “evil” which will be amended in time.
    ; cong = λ i≈j → record
      {to = record {_⟨$⟩_ = λ x≈i → x≈i ⟨≈≈⟩ i≈j; cong = λ _ → tt}
      ; from = record {_⟨$⟩_ = λ x≈j → x≈j ⟨≈≈⟩ sym i≈j; cong = λ _ → tt}
      ; inverse-of = record
        {left-inverse-of = λ _ → tt
```

```
        ; right-inverse-of  =  λ _ → tt
        }
      }
    }
  _∈₀_  :  Carrier → List Carrier → Set (ℓ ⊔ a)
  x ∈₀ xs  =  Some₀ (setoid≈ x) xs

  _∈_  :  Carrier → List Carrier → Setoid (a ⊔ ℓ) lzero
  x ∈ xs  =  Some (setoid≈ x) xs


open import Relation.Binary using (Rel)
  -- To avoid absurd patterns that we do not use, when using _⊎-Rel_, we make this:
  -- As such, we introduce the parallel composition of heterogeneous relations.
data _∥_ {a₁ b₁ c₁ a₂ b₂ c₂ : Level}
  {A₁ : Set a₁} {B₁ : Set b₁} (_~₁_ : A₁ → B₁ → Set c₁)
  {A₂ : Set a₂} {B₂ : Set b₂} (_~₂_ : A₂ → B₂ → Set c₂)
  : A₁ ⊎ A₂ → B₁ ⊎ B₂ → Set (a₁ ⊔ b₁ ⊔ c₁ ⊔ a₂ ⊔ b₂ ⊔ c₂) where
  left  : {x : A₁} {y : B₁} (x~₁y : x ~₁ y) → (_~₁_ ∥ _~₂_) (inj₁ x) (inj₁ y)
  right : {x : A₂} {y : B₂} (x~₂y : x ~₂ y) → (_~₁_ ∥ _~₂_) (inj₂ x) (inj₂ y)

  -- Before we move on, let us mention the eliminator for this type.
[_∥_] : {a₁ b₁ c₁ a₂ b₂ c₂ ℓ : Level}
      {A₁ : Set a₁} {B₁ : Set b₁} {_~₁_ : A₁ → B₁ → Set c₁}
      {A₂ : Set a₂} {B₂ : Set b₂} {_~₂_ : A₂ → B₂ → Set c₂}
  →
      {Z : Set ℓ}
      (F : {a : A₁} {b : B₁} → a ~₁ b → Z)
      (G : {a : A₂} {b : B₂} → a ~₂ b → Z)
  →
      {x : A₁ ⊎ A₂} {y : B₁ ⊎ B₂}
  → (_~₁_ ∥ _~₂_) x y → Z
[ F ∥ G ] (left x~y)  =  F x~y
[ F ∥ G ] (right x~y)  =  G x~y

  -- If the argument relations are symmetric then so is their parallel composition.
∥-sym : {a a′ c c′ : Level} {A : Set a} {_~_ : A → A → Set c}
  {A′ : Set a′} {_~′_ : A′ → A′ → Set c′}
  (sym₁ : {x y : A} → x ~ y → y ~ x) (sym₂ : {x y : A′} → x ~′ y → y ~′ x)
  {x y : A ⊎ A′}
  →
    (_~_ ∥ _~′_) x y → (_~_ ∥ _~′_) y x
∥-sym sym₁ sym₂ (left x~y)    =  left (sym₁ x~y)
∥-sym sym₁ sym₂ (right x~y)  =  right (sym₂ x~y)
  --
  -- ought to be just: [ left ∘ sym₁ ∥ right ∘ sym₂ ]
  --
infix 999 _⊎⊎_
_⊎⊎_  :  {i₁ i₂ k₁ k₂ : Level} → Setoid i₁ k₁ → Setoid i₂ k₂ → Setoid (i₁ ⊔ i₂) (i₁ ⊔ i₂ ⊔ k₁ ⊔ k₂)
A ⊎⊎ B  =  record
  {Carrier  =  A₀ ⊎ B₀
  ; _≈_  =  ≈₁ ∥ ≈₂
  ; isEquivalence  =  record
    {refl    =  λ {{inj₁ x} → left refl₁; {inj₂ x} → right refl₂}
    ; sym    =  λ {(left eq) → left (sym₁ eq); (right eq) → right (sym₂ eq)}
                 -- ought to be writable as [ left ∘ sym₁ ∥ right ∘ sym₂ ]
    ; trans  =  λ {(left eq) (left       eqq) → left (trans₁ eq eqq)
                 ; (right eq) (right eqq) → right (trans₂ eq eqq)
                 }
```

```
      }
    }
    where
      open Setoid A renaming (Carrier to A₀; _≈_ to ≈₁; refl to refl₁; sym to sym₁; trans to trans₁)
      open Setoid B renaming (Carrier to B₀; _≈_ to ≈₂; refl to refl₂; sym to sym₂; trans to trans₂)


⊎⊎-comm : {a b aℓ bℓ : Level} {A : Setoid a aℓ} {B : Setoid b bℓ} → A ⊎⊎ B ≅ B ⊎⊎ A
⊎⊎-comm {A = A} {B} = record
  {to          = record {_⟨$⟩_ = swap₊; cong = swap-on-∥}
  ; from       = record {_⟨$⟩_ = swap₊; cong = swap-on-∥′}
  ; inverse-of = record {left-inverse-of = swap²≈∥≈id; right-inverse-of = swap²≈∥≈id′}
  }
  where
    open Setoid A renaming (Carrier to A₀; _≈_ to ≈₁; refl to refl₁)
    open Setoid B renaming (Carrier to B₀; _≈_ to ≈₂; refl to refl₂)
    swap-on-∥ : {i j : A₀ ⊎ B₀} → (≈₁ ∥ ≈₂) i j → (≈₂ ∥ ≈₁) (swap₊ i) (swap₊ j)
    swap-on-∥ (left x∼₁y)   = right x∼₁y
    swap-on-∥ (right x∼₂y) = left x∼₂y
    swap²≈∥≈id : (z : A₀ ⊎ B₀) → (≈₁ ∥ ≈₂) (swap₊ (swap₊ z)) z
    swap²≈∥≈id (inj₁ _) = left refl₁
    swap²≈∥≈id (inj₂ _) = right refl₂
     {-Tried to obtain the following via ∥-sym ... -}
    swap-on-∥′ : {i j : B₀ ⊎ A₀} → (≈₂ ∥ ≈₁) i j → (≈₁ ∥ ≈₂) (swap₊ i) (swap₊ j)
    swap-on-∥′ (left x˜y) = right x˜y
    swap-on-∥′ (right x˜y) = left x˜y
    swap²≈∥≈id′ : (z : B₀ ⊎ A₀) → (≈₂ ∥ ≈₁) (swap₊ (swap₊ z)) z
    swap²≈∥≈id′ (inj₁ _) = left   refl₂
    swap²≈∥≈id′ (inj₂ _) = right refl₁


module _ {a ℓa : Level} {A : Setoid a ℓa} {P : A ⟶ SSetoid ℓa ℓa} where
  ++≅ : {xs ys : List (Setoid.Carrier A)} → (Some P xs ⊎⊎ Some P ys) ≅ Some P (xs + ys)
  ++≅ {xs} {ys} = record
    {to = record {_⟨$⟩_ = ⊎→++; cong = ⊎→++-cong}
    ; from = record {_⟨$⟩_ = ++→⊎ xs; cong = {! ++→⊎-cong xs {ys} !}}
    ; inverse-of = record
      {left-inverse-of = {! ++→⊎∘⊎→++≅id xs !}
      ; right-inverse-of = {! ⊎→++∘++→⊎≅id xs !}
      }
    }
    where
      _∼_ = _∼S_ P
        -- "ealier"
      ⊎→ˡ : ∀ {ws zs} → Some₀ P ws → Some₀ P (ws + zs)
      ⊎→ˡ (here p) = here p
      ⊎→ˡ (there p) = there (⊎→ˡ p)
        -- "later"
      ⊎→ʳ : ∀ xs {ys} → Some₀ P ys → Some₀ P (xs + ys)
      ⊎→ʳ [] p = p
      ⊎→ʳ (x :: xs₁) p = there (⊎→ʳ xs₁ p)
      ⊎→++ : ∀ {zs ws} → (Some₀ P zs ⊎ Some₀ P ws) → Some₀ P (zs + ws)
      ⊎→++ (inj₁ x) = ⊎→ˡ x
      ⊎→++ {zs} (inj₂ y) = ⊎→ʳ zs y
```

```
    ++→⊎ : ∀ xs {ys} → Some₀ P (xs ++ ys) → Some₀ P xs ⊎ Some₀ P ys
    ++→⊎ [] p = inj₂ p
    ++→⊎ (x ∷ l) (here p) = inj₁ (here p)
    ++→⊎ (x ∷ l) (there p) = (there ⊎₁ id₀) (++→⊎ l p)
      -- all of the following may need to change
    ⊎→++-cong : {a b : Some₀ P xs ⊎ Some₀ P ys} → (_∼_ ∥ _∼_) a b → ⊎→++ a ∼ ⊎→++ b
    ⊎→++-cong (left x₁∼x₂)   = {!!}
    ⊎→++-cong (right y₁∼y₂) = {!!}
    ++→⊎-cong : ∀ ws {zs} {a b : Some₀ P (ws ++ zs)} → a ≡ b → (_≡_ ∥ _≡_) (++→⊎ ws a) (++→⊎ ws b)
    ++→⊎-cong [] ≡.refl = right ≡.refl
    ++→⊎-cong (x ∷ xs) {a = here px} ≡.refl = left ≡.refl
    ++→⊎-cong (x ∷ xs) {a = there pxs} ≡.refl with ++→⊎ xs pxs | ++→⊎-cong xs {a = pxs} ≡.refl
    ...| inj₁ _ | left ≡.refl          = left  ≡.refl
    ...| inj₂ _ | right ≡.refl         = right ≡.refl
    ++→⊎∘⊎→++≅id : ∀ zs {ws} → (pf : Some₀ P zs ⊎ Some₀ P ws) → (_≡_ ∥ _≡_) (++→⊎ zs (⊎→++ pf)) pf
    ++→⊎∘⊎→++≅id [] (inj₁ ())
    ++→⊎∘⊎→++≅id [] (inj₂ _) = right ≡.refl
    ++→⊎∘⊎→++≅id (z ∷ zs) (inj₁ (here p)) = left ≡.refl
    ++→⊎∘⊎→++≅id (z ∷ zs) {ws} (inj₁ (there p)) with ++→⊎ zs {ws} (⊎→++ (inj₁ p)) | ++→⊎∘⊎→++≅id zs {ws} (inj₁ p)
    ... | inj₁ pp | left pp≡p = left (≡.cong there pp≡p)
    ++→⊎∘⊎→++≅id (z ∷ zs) {ws} (inj₂ p) with ++→⊎ zs {ws} (⊎→++ {zs} (inj₂ p)) | ++→⊎∘⊎→++≅id zs (inj₂ p)
    ... | inj₂ pp | right pp≡p = right pp≡p
    ⊎→++∘++→⊎≅id : ∀ zs {ws} → (x : Some₀ P (zs ++ ws)) → ⊎→++ {zs} {ws} (++→⊎ zs x) ≡ x
    ⊎→++∘++→⊎≅id [] x = ≡.refl
    ⊎→++∘++→⊎≅id (x ∷ zs) (here p) = ≡.refl
    ⊎→++∘++→⊎≅id (x ∷ zs) (there p) with ++→⊎ zs p | ⊎→++∘++→⊎≅id zs p
    ... | inj₁ y | ≡.refl = ≡.refl
    ... | inj₂ y | ≡.refl = ≡.refl


⊥⊥ : ∀ {a ℓa} → Setoid a ℓa
⊥⊥ {a} {ℓa} = record
  {Carrier = ⊥
  ; _≈_ = λ _ _ → ⊤
  ; isEquivalence = record {refl = tt; sym = λ _ → tt; trans = λ _ _ → tt}
  }


module _ {a ℓa : Level} {A : Setoid a ℓa} {P : A ⟶ SSetoid ℓa ℓa} where
  ⊥≅Some[] : ⊥⊥ {a} {ℓa} ≅ Some P []
  ⊥≅Some[] = record
    {to         = record {_⟨$⟩_ = λ {()}; cong = λ {{()}}}
    ; from       = record {_⟨$⟩_ = λ {()}; cong = λ {{()}}}
    ; inverse-of = record {left-inverse-of = λ _ → tt; right-inverse-of = λ {()}}
    }


map≅ : ∀ {a ℓa} {A B : Setoid a ℓa} {P : B ⟶ SSetoid ℓa ℓa} {f : A ⟶ B} {xs : List (Setoid.Carrier A)} →
  Some (P ∘ f) xs ≅ Some P (map (_⟨$⟩_ f) xs)
map≅ {A = A} {B} {P} {f} = record
  {to = record {_⟨$⟩_ = map⁺; cong = {!!}}
  ; from = record {_⟨$⟩_ = map⁻; cong = {!!}}
  ; inverse-of = record {left-inverse-of = map⁻∘map⁺; right-inverse-of = map⁺∘map⁻}}
  where
  g = _⟨$⟩_ f
  A₀ = Setoid.Carrier A
```

```
  _~_ = _~S_ P
  map⁺ : {xs : List A₀} → Some₀ (P ∘ f) xs → Some₀ P (map g xs)
  map⁺ (here p) = here p
  map⁺ (there p) = there $ map⁺ p
  map⁻ : {xs : List A₀} → Some₀ P (map g xs) → Some₀ (P ∘ f) xs
  map⁻ {[]} ()
  map⁻ {x :: xs} (here p) = here p
  map⁻ {x :: xs} (there p) = there (map⁻ {xs = xs} p)
  map⁺∘map⁻ : {xs : List A₀} → (p : Some₀ P (map g xs)) → map⁺ (map⁻ p) ~ p
  map⁺∘map⁻ {[]} ()
  map⁺∘map⁻ {x :: xs} (here p) = ≡.refl
  map⁺∘map⁻ {x :: xs} (there p) = ≡.cong suc (map⁺∘map⁻ p)
  map⁻∘map⁺ : {xs : List A₀} → (p : Some₀ (P ∘ f) xs) → let _~₂_ = _~S_ (P ∘ f) in map⁻ (map⁺ p) ~₂ p
  map⁻∘map⁺ {[]} ()
  map⁻∘map⁺ {x :: xs} (here p) = ≡.refl
  map⁻∘map⁺ {x :: xs} (there p) = ≡.cong suc (map⁻∘map⁺ p)
```

This isn't quite the full-powered cong, but is all we need.

```
module _ {a ℓa : Level} {A : Setoid a ℓa} {P : A ⟶ SSetoid ℓa ℓa} {xs : List (Setoid.Carrier A)} where
open Membership A
open Setoid A
private P₀ = λ e → Setoid.Carrier (Π._⟨$⟩_ P e)
ΣP-Setoid : Setoid (ℓa ⊔ a) ℓa
ΣP-Setoid = record
  {Carrier = Σ Carrier (λ x → (x ∈₀ xs) × P₀ x)
  ; _≈_ = λ {(a , a∈xs , Pa) (b , b∈xs , Pb) → (a ≈ b) × toℕ (setoid≈ a) a∈xs ≡ toℕ (setoid≈ b) b∈xs × ((Π._⟨$⟩_ P a) ≅ (Π._⟨$⟩_ P b))
  ; isEquivalence = record {refl = {!!}; sym = {!!}; trans = {!!}}}
find : ∀ {ys} → Some₀ P ys → ∃ (λ x → (x ∈₀ ys) × P₀ x)
find {[]} ()
find {x :: xs} (here p) = x , here (Setoid.refl A) , p
find {x :: xs} (there p) =
  let pos = find p in proj₁ pos , there (proj₁ (proj₂ pos)) , proj₂ (proj₂ pos)
lose : ∀ {ys} → Σ Carrier (λ x → x ∈₀ ys × P₀ x) → Some₀ P ys
lose (x , here px , Px) = here (_≅_.to (Π.cong P px) Π.⟨$⟩ Px)
lose (x , there x∈xs , Px) = there (lose (x , x∈xs , Px))
ΣP-Some : Some P xs ≅ ΣP-Setoid
ΣP-Some = record
  {to = record {_⟨$⟩_ = find {xs}; cong = {!!}}
  ; from = record {_⟨$⟩_ = lose; cong = lose-cong}
  ; inverse-of = record
    {left-inverse-of = left-inv
    ; right-inverse-of = {!!}
    }
  }
  where
  _~_ = _~S_ P
  lose-cong : ∀ {ys : List Carrier} {a b : Σ Carrier (λ x → x ∈₀ ys × P₀ x)} → let i = proj₁ a in let j = proj₁ b in
    let i∈ys = proj₁ (proj₂ a) in let j∈ys = proj₁ (proj₂ b) in
    i ≈ j × toℕ (setoid≈ i) i∈ys ≡ toℕ (setoid≈ j) j∈ys × ((Π._⟨$⟩_ P i) ≅ (Π._⟨$⟩_ P j)) → lose {ys} a ~ lose b
  lose-cong {_} {a₁ , here {x} px , Pa} {b , here px₁ , Pb} (i≈j , _ , Pi≅Pj) = ≡.refl
  lose-cong {_} {a₁ , here px , Pa} {b , there b∈xs , Pb} (i≈j , () , Pi≅Pj)
  lose-cong {_} {a₁ , there a∈xs , Pa} {b , here px , Pb} (i≈j , () , Pi≅Pj)
  lose-cong {_} {a₁ , there a∈xs , Pa} {b , there b∈xs , Pb} (i≈j , xx , Pi≅Pj) =
    ≡.cong suc (lose-cong {a = a₁ , a∈xs , Pa} {b , b∈xs , Pb} (i≈j , suc-inj xx , Pi≅Pj))
```

```
left-inv : ∀ {ys} (x : Some₀ P ys) → toℕ P (lose (find x)) ≡ toℕ P x
left-inv (here px) = ≡.refl
left-inv (there x₁) = ≡.cong suc (left-inv x₁)
```

```
module _ {a ℓa : Level} {A : Setoid a ℓa} {P : A ⟶ SSetoid ℓa ℓa} where
open Membership A
open Setoid A
private P₀ = λ e → Setoid.Carrier (Π._⟨$⟩_ P e)
Some-cong : {xs₁ xs₂ : List Carrier} →
  (∀ {x} → (x ∈ xs₁) ≅ (x ∈ xs₂)) →
  Some P xs₁ ≅ Some P xs₂
Some-cong {xs₁} {xs₂} list-rel = record
  {to         = record {_⟨$⟩_ = xs₁→xs₂ list-rel; cong = {!!}}
  ; from      = record {_⟨$⟩_ = xs₁→xs₂ (≅-sym list-rel); cong = {!!}}
  ; inverse-of = record {left-inverse-of = left-inv list-rel; right-inverse-of = {!!}}
  }
  where
  copy : ∀ {x} {ys} → x ∈₀ ys → P₀ x → Some₀ P ys
  copy (here p) pf = here (_≅_.to (Π.cong P p) ⟨$⟩ pf)
  copy (there p) pf = there (copy p pf)
  xs₁→xs₂ : ∀ {xs ys} → (∀ {x} → (x ∈ xs) ≅ (x ∈ ys)) → Some₀ P xs → Some₀ P ys
  xs₁→xs₂ {[]} _ ()
  xs₁→xs₂ {x :: xs} rel (here p) = copy (_≅_.to rel ⟨$⟩ here (Setoid.refl A)) p
  xs₁→xs₂ {x :: xs} {ys} rel (there p) =
    let pos = find p in copy (_≅_.to rel ⟨$⟩ there (proj₁ (proj₂ pos))) (proj₂ (proj₂ pos))
  left-inv : ∀ {xs ys} → (rel : ∀ {x} → (x ∈ xs) ≅ (x ∈ ys)) → (∀ y → xs₁→xs₂ (≅-sym rel) (xs₁→xs₂ rel y) ≡ y)
  left-inv {[]} rel ()
  left-inv {x :: xs} rel (here p) with _≅_.to rel ⟨$⟩ here refl | inspect (_⟨$⟩_ (_≅_.to rel)) (here refl)
  ... | here pp | [ eq ] = {!!}
  ... | there qq | [ eq ] = {!!}
  left-inv {x :: xs} rel (there p) = {!!}
```

11 Conclusion and Outlook

$\boxed{???}$