

Theories and Data Structures

Jacques Carette, Musa Al-hassy, Wolfram Kahl

June 15, 2017

Abstract

We aim to show how common data-structures naturally arise from elementary mathematical theories.

In particular, we answer the following questions:

- Why do lists pop-up more frequently to the average programmer than, say, their duals: bags?
- More simply, why do unit and empty types occur so naturally? What about enumerations/sums and records/products?
- Why is it that dependent sums and products do not pop-up explicitly to the average programmer? They arise naturally all the time as tuples and as classes.
- How do we get the usual toolbox of functions and helpful combinators for a particular data type? Are they “built into” the type?
- Is it that the average programmer works in the category of classical Sets, with functions and propositional equality? Does this result in some “free constructions” not easily made computable since mathematicians usually work in the category of Setoids but tend to quotient to arrive in **Sets**? —where quotienting is not computably feasible, in **Sets** at-least; and why is that?

???

This research is supported by the National Science and Engineering Research Council (NSERC), Canada

Contents

1	Introduction	4
2	Overview	4
3	Obtaining Forgetful Functors	4
4	Equality Combinators	5
4.1	Propositional Equality	5
4.2	Function Extensionality	6
4.3	Equiv	6
4.4	Making <code>symmetry</code> calls less intrusive	6
5	Properties of Sums and Products	7
5.1	Generalised Bot and Top	7
5.2	Sums	8
5.3	Products	8
6	Two Sorted Structures	9
6.1	Definitions	9
6.2	Category and Forgetful Functors	10
6.3	Free and CoFree	10
6.4	Adjunction Proofs	11
6.5	Merging is adjoint to duplication	12
6.6	Duplication also has a left adjoint	13
7	Binary Heterogeneous Relations — [MA:] <i>What named data structure do these correspond to in programming?</i> []	13
7.1	Definitions	14
7.2	Category and Forgetful Functors	14
7.3	Free and CoFree Functors	15
7.4	[???]	19
8	Pointed Algebras: Nullable Types	20
8.1	Definition	20
8.2	Category and Forgetful Functors	21
8.3	A Free Construction	21
9	SetoidSetoid	22
10	Some	23
10.1	<code>Some₀</code>	23
10.2	Membership module	24

10.3 Parallel Composition	25
10.4 $\text{⋈}\text{⋈}$ -comm	26
10.5 $++\cong : \dots \rightarrow (\text{Some } P \text{ } xs \text{ } \text{⋈}\text{⋈} \text{ } \text{Some } P \text{ } ys) \cong \text{Some } P \text{ } (xs + ys)$	26
10.6 Bottom as a setoid	28
10.7 $\text{map}\cong : \dots \rightarrow \text{Some } (P \circ f) \text{ } xs \cong \text{Some } P \text{ } (\text{map } (_ \langle \$ \rangle _) f) \text{ } xs$	28
10.8 Some-cong and holes	30
11 Conclusion and Outlook	32

1 Introduction

???

2 Overview

???

The Agda source code for this development is available on-line at the following URL:

<https://github.com/JacquesCarette/TheoriesAndDataStructures>

3 Obtaining Forgetful Functors

We aim to realise a “toolkit” for an data-structure by considering a free construction and proving it adjoint to a forgetful functor. Since the majority of our theories are built on the category **Set**, we begin by making a helper method to produce the forgetful functors from as little information as needed about the mathematical structure being studied.

Indeed, it is a common scenario where we have an algebraic structure with a single carrier set and we are interested in the categories of such structures along with functions preserving the structure.

We consider a type of “algebras” built upon the category of **Sets** —in that, every algebra has a carrier set and every homomorphism is essentially a function between carrier sets where the composition of homomorphisms is essentially the composition of functions and the identity homomorphism is essentially the identity function.

Such algebras constitute a category from which we obtain a method to Forgetful functor builder for single-sorted algebras to **Sets**.

```

module Forget where
open import Level
open import Categories.Category using (Category)
open import Categories.Functor using (Functor)
open import Categories.Agda using (Sets)
open import Function2
open import Function
open import EqualityCombinators

```

[MA: For one reason or another, the module head is not making the imports smaller.]

A **OneSortedAlg** is essentially the details of a forgetful functor from some category to **Sets**,

```

record OneSortedAlg (ℓ : Level) : Set (suc (suc ℓ)) where
  field
    Alg      : Set (suc ℓ)
    Carrier  : Alg → Set ℓ
    Hom      : Alg → Alg → Set ℓ
    mor      : {A B : Alg} → Hom A B → (Carrier A → Carrier B)
    comp     : {A B C : Alg} → Hom B C → Hom A B → Hom A C
    .comp-is-o : {A B C : Alg} {g : Hom B C} {f : Hom A B} → mor (comp g f) ≐ mor g ∘ mor f
    Id       : {A : Alg} → Hom A A
    .Id-is-id : {A : Alg} → mor (Id {A}) ≐ id

```

The aforementioned claim that algebras and their structure preserving morphisms form a category can be realised due to the coherency conditions we requested viz the morphism operation on homomorphisms is functorial.

```

open import Relation.Binary.SetoidReasoning
oneSortedCategory : (ℓ : Level) → OneSortedAlg ℓ → Category (suc ℓ) ℓ ℓ
oneSortedCategory ℓ A = record
  { Obj      = Alg
  ; _⇒_      = Hom
  ; _≡_      = λ F G → mor F ≐ mor G
  ; id       = Id
  ; _o_      = comp
  ; assoc    = λ {A B C D} {F} {G} {H} → begin⟨ ≐-setoid (Carrier A) (Carrier D) ⟩
    mor (comp (comp H G) F) ≈⟨ comp-is-o ⟩
    mor (comp H G) o mor F   ≈⟨ o-≐-cong1 _ comp-is-o ⟩
    mor H o mor G o mor F    ≈⟨ o-≐-cong2 (mor H) comp-is-o ⟩
    mor H o mor (comp G F)   ≈⟨ comp-is-o ⟩
    mor (comp H (comp G F)) ■
  ; identityl = λ {f = f} → comp-is-o ⟨ ≐ ⟩ Id-is-id o mor f
  ; identityr = λ {f = f} → comp-is-o ⟨ ≐ ⟩ ≡.cong (mor f) o Id-is-id
  ; equiv     = record { IsEquivalence ≐-isEquivalence }
  ; o-resp-≡  = λ f≈h g≈k → comp-is-o ⟨ ≐ ⟩ o-resp-≐ f≈h g≈k ⟨ ≐ ⟩ ≐-sym comp-is-o
  }
where open OneSortedAlg A; open import Relation.Binary using (IsEquivalence)

```

The fact that the algebras are built on the category of sets is captured by the existence of a forgetful functor.

```

mkForgetful : (ℓ : Level) (A : OneSortedAlg ℓ) → Functor (oneSortedCategory ℓ A) (Sets ℓ)
mkForgetful ℓ A = record
  { F0      = Carrier
  ; F1      = mor
  ; identity  = Id-is-id $i
  ; homomorphism = comp-is-o $i
  ; F-resp-≡  = _$i
  }
where open OneSortedAlg A

```

That is, the constituents of a `OneSortedAlgebra` suffice to produce a category and a so-called presheaf as well.

4 Equality Combinators

Here we export all equality related concepts, including those for propositional and function extensional equality.

```

module EqualityCombinators where
open import Level

```

4.1 Propositional Equality

We use one of Agda’s features to qualify all propositional equality properties by “≡.” for the sake of clarity and to avoid name clashes with similar other properties.

```

import Relation.Binary.PropositionalEquality
module ≡ = Relation.Binary.PropositionalEquality
open ≡ using ( _≡_ ) public

```

We also provide two handy-dandy combinators for common uses of transitivity proofs.

```

_⟨≡≡⟩_ = ≡.trans
_⟨≡≡⟩_ : {a : Level} {A : Set a} {x y z : A} → x ≡ y → z ≡ y → x ≡ z
x≈y ⟨≡≡⟩ z≈y = x≈y ⟨≡≡⟩ ≡.sym z≈y

```

4.2 Function Extensionality

We bring into scope pointwise equality, $_ \dot{=} _$, and provide a proof that it constitutes an equivalence relation—where the source and target of the functions being compared are left implicit.

```

open ≡ using () renaming ( _→-setoid_ to ≡-setoid; _≡_ to ≡- ) public
open import Relation.Binary using (IsEquivalence; Setoid)
module _ {a b : Level} {A : Set a} {B : Set b} where
  ≡-isEquivalence : IsEquivalence ( _≡_ {A = A} {B} )
  ≡-isEquivalence = record { Setoid (≡-setoid A B) }
  open IsEquivalence ≡-isEquivalence public
  renaming ( refl to ≡-refl; sym to ≡-sym; trans to ≡-trans )
  open import Equiv public using ( o- resp-≡ ) -- To do: port this over here!
  renaming ( cong∘ to o-≡-cong₂; cong∘ to o-≡-cong₁ )
infixr 5 _⟨≡≡⟩_
_⟨≡≡⟩_ = ≡-trans

```

Note that the precedence of this last operator is lower than that of function composition so as to avoid superfluous parenthesis.

4.3 Equiv

We form some combinators for HoTT like reasoning.

```

cong₂D : ∀ {a b c} {A : Set a} {B : A → Set b} {C : Set c}
  (f : (x : A) → B x → C)
  → {x₁ x₂ : A} {y₁ : B x₁} {y₂ : B x₂}
  → (x₂≡x₁ : x₂ ≡ x₁) → ≡.subst B x₂≡x₁ y₂ ≡ y₁ → f x₁ y₁ ≡ f x₂ y₂
cong₂D f ≡.refl ≡.refl = ≡.refl
open import Equiv public using ( _≃_ ; id≃ ; sym≃ ; trans≃ ; qinv )
infix 3 _□_
infixr 2 _≃⟨_⟩_
_≃⟨_⟩_ : {x y z : Level} (X : Set x) {Y : Set y} {Z : Set z}
  → X ≃ Y → Y ≃ Z → X ≃ Z
X ≃⟨ X≃Y ⟩ Y≃Z = trans≃ X≃Y Y≃Z
_□_ : {x : Level} (X : Set x) → X ≃ X
X□ = id≃

```

[MA: Consider moving pertinent material here from *Equiv.lagda* at the end.]

4.4 Making symmetry calls less intrusive

It is common that we want to use an equality within a calculation as a right-to-left rewrite rule which is accomplished by utilizing its symmetry property. We simplify this rendition, thereby saving an explicit call and parenthesis in-favour of a less hinder-some notation.

Among other places, I want to use this combinator in module `Forget`’s proof of associativity for `oneSortedCategory`

```

module _ {c l : Level} {S : Setoid c l} where
  open import Relation.Binary.SetoidReasoning using ( _≈⟨_⟩_ )
  open import Relation.Binary.EqReasoning using ( _IsRelatedTo_ )
  open Setoid S
  infixr 2 _≈⟨_⟩_
  _≈⟨_⟩_ : ∀ (x {y z} : Carrier) → y ≈ x → _IsRelatedTo_ S y z → _IsRelatedTo_ S x z
  x ≈⟨ y≈x ⟩ y≈z = x ≈⟨ sym y≈x ⟩ y≈z

```

A host of similar such combinators can be found within the RATH-Agda library.

5 Properties of Sums and Products

This module is for those domain-ubiquitous properties that, disappointingly, we could not locate in the standard library. —The standard library needs some sort of “table of contents *with* subsection” to make it easier to know of what is available.

This module re-exports (some of) the contents of the standard library’s `Data.Product` and `Data.Sum`.

```

module DataProperties where
  open import Level renaming (suc to lsuc; zero to lzero)
  open import Function using (id; _◦_; const)
  open import EqualityCombinators
  open import Data.Product public using ( _×_ ; proj1; proj2; Σ; _,_ ; swap; uncurry ) renaming (map to _×1_ ; <_,_> to ⟨_,_⟩ )
  open import Data.Sum public using ( inj1; inj2; [_,_] ) renaming (map to _⊔1_ )
  open import Data.Nat using (ℕ; zero; suc)

```

Precedence Levels

The standard library assigns precedence level of 1 for the infix operator `_⊔_`, which is rather odd since infix operators ought to have higher precedence than equality combinators, yet the standard library assigns `_≈⟨_⟩_` a precedence level of 2. The usage of these two —e.g. in `CommMonoid.lagda`— causes an annoying number of parentheses and so we reassign the level of the infix operator to avoid such a situation.

```

infixr 3 _⊔_
_⊔_ = Data.Sum._⊔_

```

5.1 Generalised Bot and Top

To avoid a flurry of lift’s, and for the sake of clarity, we define level-polymorphic empty and unit types.

```

open import Level
data ⊥ {ℓ : Level} : Set ℓ where
  ⊥-elim : {a ℓ : Level} {A : Set a} → ⊥ {ℓ} → A
  ⊥-elim ()
record ⊤ {ℓ : Level} : Set ℓ where
  constructor tt

```

5.2 Sums

Just as $_ \sqcup _$ takes types to types, its “map” variant $_ \sqcup_1 _$ takes functions to functions and is a functorial congruence: It preserves identity, distributes over composition, and preserves extensional equality.

```

 $\sqcup$ -id : {a b : Level} {A : Set a} {B : Set b} → id  $\sqcup_1$  id  $\doteq$  id {A = A  $\sqcup$  B}
 $\sqcup$ -id = [  $\doteq$ -refl ,  $\doteq$ -refl ]

 $\sqcup$ -o : {a b c a' b' c' : Level}
  {A : Set a} {A' : Set a'} {B : Set b} {B' : Set b'} {C : Set c} {C' : Set c'}
  {f : A → A'} {g : B → B'} {f' : A' → C} {g' : B' → C'}
  → (f' ∘ f)  $\sqcup_1$  (g' ∘ g)  $\doteq$  (f'  $\sqcup_1$  g') ∘ (f  $\sqcup_1$  g) -- aka “the exchange rule for sums”
 $\sqcup$ -o = [  $\doteq$ -refl ,  $\doteq$ -refl ]

 $\sqcup$ -cong : {a b c d : Level} {A : Set a} {B : Set b} {C : Set c} {D : Set d} {f f' : A → C} {g g' : B → D}
  → f  $\doteq$  f' → g  $\doteq$  g' → f  $\sqcup_1$  g  $\doteq$  f'  $\sqcup_1$  g'
 $\sqcup$ -cong f $\approx$ f' g $\approx$ g' = [ o- $\doteq$ -cong2 inj1 f $\approx$ f' , o- $\doteq$ -cong2 inj2 g $\approx$ g' ]

```

Composition post-distributes into casing,

```

o-[.] : {a b c d : Level} {A : Set a} {B : Set b} {C : Set c} {D : Set d} {f : A → C} {g : B → C} {h : C → D}
  → h ∘ [ f , g ]  $\doteq$  [ h ∘ f , h ∘ g ] -- aka “fusion”
o-[.] = [  $\doteq$ -refl ,  $\doteq$ -refl ]

```

It is common that a data-type constructor $D : \text{Set} \rightarrow \text{Set}$ allows us to extract elements of the underlying type and so we have a natural transformation $D \rightarrow \mathbf{I}$, where \mathbf{I} is the identity functor. These kind of results will occur for our other simple data-structures as well. In particular, this is the case for $D\ A = 2 \times A = A \sqcup A$:

```

from $\sqcup$  : { $\ell$  : Level} {A : Set  $\ell$ } → A  $\sqcup$  A → A
from $\sqcup$  = [ id , id ]
-- from $\sqcup$  is a natural transformation
--
from $\sqcup$ -nat : {a b : Level} {A : Set a} {B : Set b} {f : A → B} → f ∘ from $\sqcup$   $\doteq$  from $\sqcup$  ∘ (f  $\sqcup_1$  f)
from $\sqcup$ -nat = [  $\doteq$ -refl ,  $\doteq$ -refl ]
-- from $\sqcup$  is injective and so is pre-invertible,
--
from $\sqcup$ -preInverse : {a b : Level} {A : Set a} {B : Set b} → id  $\doteq$  from $\sqcup$  {A = A  $\sqcup$  B} ∘ (inj1  $\sqcup_1$  inj2)
from $\sqcup$ -preInverse = [  $\doteq$ -refl ,  $\doteq$ -refl ]

```

[MA: insert:] A brief mention about co-monads? []

5.3 Products

Dual to $\text{from}\sqcup$, a natural transformation $2 \times _ \rightarrow \mathbf{I}$, is diag , the transformation $\mathbf{I} \rightarrow _{}^2$.

```

diag : { $\ell$  : Level} {A : Set  $\ell$ } (a : A) → A × A
diag a = a , a

```

[MA: insert:] A brief mention of Haskell’s `const`, which is `diag` curried. Also something about `K` combinator?

[]

Z-style notation for sums,

```

 $\Sigma$ :• : {a b : Level} (A : Set a) (B : A → Set b) → Set (a  $\sqcup$  b)
 $\Sigma$ :• = Data.Product. $\Sigma$ 

```


infix -666 $\Sigma : \bullet$

syntax $\Sigma : \bullet A (\lambda x \rightarrow B) = \Sigma x : A \bullet B$

open import Data.Nat.Properties

suc-inj : $\forall \{i\ j\} \rightarrow \mathbb{N}.suc\ i \equiv \mathbb{N}.suc\ j \rightarrow i \equiv j$

suc-inj = cancel-+-left ($\mathbb{N}.suc\ \mathbb{N}.zero$)

or

suc-inj {0} $_ \equiv _ .refl = _ \equiv _ .refl$

suc-inj { $\mathbb{N}.suc\ i$ } $_ \equiv _ .refl = _ \equiv _ .refl$

6 Two Sorted Structures

So far we have been considering algebraic structures with only one underlying carrier set, however programmers are faced with a variety of different types at the same time, and the graph structure between them, and so we consider briefly consider two sorted structures by starting the simplest possible case: Two type and no required interaction whatsoever between them.

module Structures.TwoSorted **where**

open import Level **renaming** (suc to lsuc; zero to lzero)

open import Categories.Category **using** (Category)

open import Categories.Functor **using** (Functor)

open import Categories.Adjunction **using** (Adjunction)

open import Categories.Agda **using** (Sets)

open import Function **using** (id; $_ \circ _$; const)

open import Function2 **using** ($_ \$i$)

open import Forget

open import EqualityCombinators

open import DataProperties

6.1 Definitions

A TwoSorted type is just a pair of sets in the same universe—in the future, we may consider those in different levels.

record TwoSorted $\ell : \text{Set } (\text{lsuc } \ell)$ **where**

constructor MkTwo

field

One : $\text{Set } \ell$

Two : $\text{Set } \ell$

open TwoSorted

Unastonishingly, a morphism between such types is a pair of functions between the *multiple* underlying carriers.

record Hom $\{\ell\}$ (Src Tgt : TwoSorted ℓ) : $\text{Set } \ell$ **where**

constructor MkHom

field

one : $\text{One Src} \rightarrow \text{One Tgt}$

two : $\text{Two Src} \rightarrow \text{Two Tgt}$

open Hom

6.2 Category and Forgetful Functors

We are using pairs of object and pairs of morphisms which are known to form a category:

$\text{Twos} : (\ell : \text{Level}) \rightarrow \text{Category} (\text{Isuc } \ell) \ell \ell$

$\text{Twos } \ell = \text{record}$

```

{Obj      = TwoSorted  $\ell$ 
;  $\Rightarrow$  _ = Hom
;  $\equiv$  _   =  $\lambda F G \rightarrow \text{one } F \doteq \text{one } G \times \text{two } F \doteq \text{two } G$ 
; id       = MkHom id id
;  $\circ$  _     =  $\lambda F G \rightarrow \text{MkHom } (\text{one } F \circ \text{one } G) (\text{two } F \circ \text{two } G)$ 
; assoc    =  $\doteq\text{-refl}, \doteq\text{-refl}$ 
; identityl =  $\doteq\text{-refl}, \doteq\text{-refl}$ 
; identityr =  $\doteq\text{-refl}, \doteq\text{-refl}$ 
; equiv    = record
  { refl    =  $\doteq\text{-refl}, \doteq\text{-refl}$ 
  ; sym     =  $\lambda \{(oneEq, twoEq) \rightarrow \doteq\text{-sym oneEq}, \doteq\text{-sym twoEq}\}$ 
  ; trans   =  $\lambda \{(oneEq_1, twoEq_1) (oneEq_2, twoEq_2) \rightarrow \doteq\text{-trans oneEq}_1 \text{ oneEq}_2, \doteq\text{-trans twoEq}_1 \text{ twoEq}_2\}$ 
  }
; o-resp- $\equiv$  =  $\lambda \{(g_{\approx_1} k, g_{\approx_2} k) (f_{\approx_1} h, f_{\approx_2} h) \rightarrow \text{o-resp-} \doteq g_{\approx_1} k f_{\approx_1} h, \text{o-resp-} \doteq g_{\approx_2} k f_{\approx_2} h\}$ 
}
```

The naming **Twos** is to be consistent with the category theory library we are using, which names the category of sets and functions by **Sets**.

We can forget about the first sort or the second to arrive at our starting category and so we have two forgetful functors.

$\text{Forget} : (\ell : \text{Level}) \rightarrow \text{Functor} (\text{Twos } \ell) (\text{Sets } \ell)$

$\text{Forget } \ell = \text{record}$

```

{F0      = TwoSorted.One
; F1      = Hom.one
; identity  =  $\equiv$ .refl
; homomorphism =  $\equiv$ .refl
; F-resp- $\equiv$  =  $\lambda \{(F_{\approx_1} G, F_{\approx_2} G) \{x\} \rightarrow F_{\approx_1} G x\}$ 
}
```

$\text{Forget}^2 : (\ell : \text{Level}) \rightarrow \text{Functor} (\text{Twos } \ell) (\text{Sets } \ell)$

$\text{Forget}^2 \ell = \text{record}$

```

{F0      = TwoSorted.Two
; F1      = Hom.two
; identity  =  $\equiv$ .refl
; homomorphism =  $\equiv$ .refl
; F-resp- $\equiv$  =  $\lambda \{(F_{\approx_1} G, F_{\approx_2} G) \{x\} \rightarrow F_{\approx_2} G x\}$ 
}
```

6.3 Free and CoFree

Given a type, we can pair it with the empty type or the singleton type and so we have a free and a co-free constructions. Intuitively, the first is free since the singleton type is the smallest type we can adjoin to obtain a **Twos** object, whereas \top is the “largest” type we adjoin to obtain a **Twos** object. This is one way that the unit and empty types naturally arise.

$\text{Free} : (\ell : \text{Level}) \rightarrow \text{Functor} (\text{Sets } \ell) (\text{Twos } \ell)$

$\text{Free } \ell = \text{record}$

```

{F0      =  $\lambda A \rightarrow \text{MkTwo } A \perp$ 
```

```

;F1          = λ f → MkHom f id
;identity      = ≐-refl , ≐-refl
;homomorphism  = ≐-refl , ≐-refl
;F-resp≡      = λ f≈g → (λ x → f≈g {x}) , ≐-refl
}

Cofree : (ℓ : Level) → Functor (Sets ℓ) (Twos ℓ)
Cofree ℓ = record
  {F0          = λ A → MkTwo A ⊤
;F1          = λ f → MkHom f id
;identity      = ≐-refl , ≐-refl
;homomorphism  = ≐-refl , ≐-refl
;F-resp≡      = λ f≈g → (λ x → f≈g {x}) , ≐-refl
}

-- Dually, ( also shorter due to eta reduction )

Free2 : (ℓ : Level) → Functor (Sets ℓ) (Twos ℓ)
Free2 ℓ = record
  {F0          = MkTwo ⊥
;F1          = MkHom id
;identity      = ≐-refl , ≐-refl
;homomorphism  = ≐-refl , ≐-refl
;F-resp≡      = λ f≈g → ≐-refl , λ x → f≈g {x}
}

Cofree2 : (ℓ : Level) → Functor (Sets ℓ) (Twos ℓ)
Cofree2 ℓ = record
  {F0          = MkTwo ⊤
;F1          = MkHom id
;identity      = ≐-refl , ≐-refl
;homomorphism  = ≐-refl , ≐-refl
;F-resp≡      = λ f≈g → ≐-refl , λ x → f≈g {x}
}

```

6.4 Adjunction Proofs

Now for the actual proofs that the `Free` and `Cofree` functors are deserving of their names.

```

Left : (ℓ : Level) → Adjunction (Free ℓ) (Forget ℓ)
Left ℓ = record
  {unit = record
    {η = λ _ → id
;commute = λ _ → ≡.refl
}
;counit = record
  {η = λ _ → MkHom id (λ {()})
;commute = λ f → ≐-refl , (λ {()})
}
;zig = ≐-refl , (λ {()})
;zag = ≡.refl
}

Right : (ℓ : Level) → Adjunction (Forget ℓ) (Cofree ℓ)
Right ℓ = record
  {unit = record
    {η = λ _ → MkHom id (λ _ → tt)
;commute = λ _ → ≐-refl , ≐-refl
}
}

```

```

; counit = record {η = λ _ → id; commute = λ _ → ≡.refl}
; zig    = ≡.refl
; zag    = ≡-refl , λ {tt → ≡.refl}
}
-- Dually,
Left2 : (ℓ : Level) → Adjunction (Free2 ℓ) (Forget2 ℓ)
Left2 ℓ = record
{ unit = record
  { η = λ _ → id
    ; commute = λ _ → ≡.refl
  }
; counit = record
  { η = λ _ → MkHom (λ {()}) id
    ; commute = λ f → (λ {()}) , ≡-refl
  }
; zig = (λ {()}) , ≡-refl
; zag = ≡.refl
}
Right2 : (ℓ : Level) → Adjunction (Forget2 ℓ) (Cofree2 ℓ)
Right2 ℓ = record
{ unit = record
  { η = λ _ → MkHom (λ _ → tt) id
    ; commute = λ _ → ≡-refl , ≡-refl
  }
; counit = record {η = λ _ → id; commute = λ _ → ≡.refl}
; zig    = ≡.refl
; zag    = (λ {tt → ≡.refl}) , ≡-refl
}

```

6.5 Merging is adjoint to duplication

The category of sets contains products and so `TwoSorted` algebras can be represented there and, moreover, this is adjoint to duplicating a type to obtain a `TwoSorted` algebra.

-- The category of Sets has products and so the `TwoSorted` type can be reified there.

```

Merge : (ℓ : Level) → Functor (Twos ℓ) (Sets ℓ)
Merge ℓ = record
{ F0      = λ S → One S × Two S
; F1      = λ F → one F ×1 two F
; identity = ≡.refl
; homomorphism = ≡.refl
; F-resp≡ = λ {(F≈1 G , F≈2 G) {x , y} → ≡.cong2 _ , _ (F≈1 G x) (F≈2 G y)}
}

```

-- Every set gives rise to its square as a `TwoSorted` type.

```

Dup : (ℓ : Level) → Functor (Sets ℓ) (Twos ℓ)
Dup ℓ = record
{ F0      = λ A → MkTwo A A
; F1      = λ f → MkHom f f
; identity = ≡-refl , ≡-refl
; homomorphism = ≡-refl , ≡-refl
; F-resp≡ = λ F≈G → diag (λ _ → F≈G)
}

```

Then the proof that these two form the desired adjunction

```

Right2 : (ℓ : Level) → Adjunction (Dup ℓ) (Merge ℓ)
Right2 ℓ = record
  {unit    = record {η = λ _ → diag; commute = λ _ → ≡.refl}
  ;counit  = record {η = λ _ → MkHom proj1 proj2; commute = λ _ → ≡-refl , ≡-refl}
  ;zig     = ≡-refl , ≡-refl
  ;zag     = ≡.refl
  }

```

6.6 Duplication also has a left adjoint

The category of sets admits sums and so an alternative is to represent a `TwoSorted` algebra as a sum, and moreover this is adjoint to the aforementioned duplication functor.

```

Choice : (ℓ : Level) → Functor (Twos ℓ) (Sets ℓ)
Choice ℓ = record
  {F0      = λ S → One S ⊔ Two S
  ;F1      = λ F → one F ⊔1 two F
  ;identity  = ⊔-id $i
  ;homomorphism = λ {x = x} → ⊔-o x
  ;F-resp≡ = λ F≈G {x} → uncurry ⊔-cong F≈G x
  }
Left2 : (ℓ : Level) → Adjunction (Choice ℓ) (Dup ℓ)
Left2 ℓ = record
  {unit      = record {η = λ _ → MkHom inj1 inj2; commute = λ _ → ≡-refl , ≡-refl}
  ;counit    = record {η = λ _ → from⊔; commute = λ _ {x} → (≡.sym ∘ from⊔-nat) x}
  ;zig       = λ { {- } } {x} → from⊔-preInverse x
  ;zag       = ≡-refl , ≡-refl
  }

```

7 Binary Heterogeneous Relations — [MA: What named data structure do these correspond to in programming?]

We consider two sorted algebras endowed with a binary heterogeneous relation. An example of such a structure is a graph, or network, which has a sort for edges and a sort for nodes and an incidence relation.

```

module Structures.Rel where
open import Level renaming (suc to lsuc; zero to lzero; _⊔_ to _⊔_)
open import Categories.Category using (Category)
open import Categories.Functor using (Functor)
open import Categories.Adjunction using (Adjunction)
open import Categories.Agda using (Sets)
open import Function using (id; _o_; const)
open import Function2 using (_$i)
open import Forget
open import EqualityCombinators
open import DataProperties
open import Structures.TwoSorted using (TwoSorted; Twos; MkTwo) renaming (Hom to TwoHom; MkHom to MkTwoHom)

```

7.1 Definitions

We define the structure involved, along with a notational convenience:

```
record HetroRel  $\ell \ell'$  : Set (Isuc ( $\ell \sqcup \ell'$ )) where
  constructor MkHRel
  field
    One : Set  $\ell$ 
    Two : Set  $\ell$ 
    Rel : One  $\rightarrow$  Two  $\rightarrow$  Set  $\ell'$ 
open HetroRel
relOp = HetroRel.Rel
syntax relOp A  $\times$  y = x  $\langle$  A  $\rangle$  y
```

Then define the strcture-preserving operations,

```
record Hom { $\ell \ell'$ } (Src Tgt : HetroRel  $\ell \ell'$ ) : Set ( $\ell \sqcup \ell'$ ) where
  constructor MkHom
  field
    one : One Src  $\rightarrow$  One Tgt
    two : Two Src  $\rightarrow$  Two Tgt
    shift : {x : One Src} {y : Two Src}  $\rightarrow$  x  $\langle$  Src  $\rangle$  y  $\rightarrow$  one x  $\langle$  Tgt  $\rangle$  two y
open Hom
```

7.2 Category and Forgetful Functors

That these structures form a two-sorted algebraic category can easily be witnessed.

```
Rels : ( $\ell \ell'$  : Level)  $\rightarrow$  Category (Isuc ( $\ell \sqcup \ell'$ )) ( $\ell \sqcup \ell'$ )  $\ell$ 
Rels  $\ell \ell'$  = record
  {Obj      = HetroRel  $\ell \ell'$ 
  ;  $\_ \Rightarrow \_$  = Hom
  ;  $\_ \equiv \_$     =  $\lambda$  F G  $\rightarrow$  one F  $\doteq$  one G  $\times$  two F  $\doteq$  two G
  ; id        = MkHom id id id
  ;  $\_ \circ \_$       =  $\lambda$  F G  $\rightarrow$  MkHom (one F  $\circ$  one G) (two F  $\circ$  two G) (shift F  $\circ$  shift G)
  ; assoc     =  $\doteq$ -refl ,  $\doteq$ -refl
  ; identityl =  $\doteq$ -refl ,  $\doteq$ -refl
  ; identityr =  $\doteq$ -refl ,  $\doteq$ -refl
  ; equiv     = record
    { refl =  $\doteq$ -refl ,  $\doteq$ -refl
    ; sym  =  $\lambda$  {(oneEq , twoEq)  $\rightarrow$   $\doteq$ -sym oneEq ,  $\doteq$ -sym twoEq}
    ; trans =  $\lambda$  {(oneEq1 , twoEq1) (oneEq2 , twoEq2)  $\rightarrow$   $\doteq$ -trans oneEq1 oneEq2 ,  $\doteq$ -trans twoEq1 twoEq2}
    }
  ; o-resp- $\equiv$  =  $\lambda$  {(g $\approx$ 1 k , g $\approx$ 2 k) (f $\approx$ 1 h , f $\approx$ 2 h)  $\rightarrow$  o-resp- $\doteq$  g $\approx$ 1 k f $\approx$ 1 h , o-resp- $\doteq$  g $\approx$ 2 k f $\approx$ 2 h}
  }
```

We can forget about the first sort or the second to arrive at our starting category and so we have two forgetful functors. Moreover, we can simply forget about the relation to arrive at the two-sorted category :-)

```
Forget1 : ( $\ell \ell'$  : Level)  $\rightarrow$  Functor (Rels  $\ell \ell'$ ) (Sets  $\ell$ )
Forget1  $\ell \ell'$  = record
  {F0      = HetroRel.One
  ; F1      = Hom.one
  ; identity =  $\equiv$ .refl
```

```

;homomorphism = ≡.refl
;F-resp≡ = λ {(F≈1G , F≈2G) {x} → F≈1G x}
}
Forget2 : (ℓ ℓ' : Level) → Functor (Rels ℓ ℓ') (Sets ℓ)
Forget2 ℓ ℓ' = record
  {F0          = HetroRel.Two
;F1          = Hom.two
;identity      = ≡.refl
;homomorphism = ≡.refl
;F-resp≡      = λ {(F≈1G , F≈2G) {x} → F≈2G x}
}
-- Whence, Rels is a subcategory of Twos
Forget3 : (ℓ ℓ' : Level) → Functor (Rels ℓ ℓ') (Twos ℓ)
Forget3 ℓ ℓ' = record
  {F0          = λ S → MkTwo (One S) (Two S)
;F1          = λ F → MkTwoHom (one F) (two F)
;identity      = ≡-refl , ≡-refl
;homomorphism = ≡-refl , ≡-refl
;F-resp≡      = id
}

```

7.3 Free and CoFree Functors

Given a (two)type, we can pair it with the empty type or the singleton type and so we have a free and a co-free constructions. Intuitively, the empty type denotes the empty relation which is the smallest relation and so a free construction; whereas, the singleton type denotes the “always true” relation which is the largest binary relation and so a cofree construction.

Candidate adjoints to forgetting the *first* component of a Rels

```

Free1 : (ℓ ℓ' : Level) → Functor (Sets ℓ) (Rels ℓ ℓ')
Free1 ℓ ℓ' = record
  {F0          = λ A → MkHRel A ⊥ (λ { _ } ())
;F1          = λ f → MkHom f id (λ { {y = ()} })
;identity      = ≡-refl , ≡-refl
;homomorphism = ≡-refl , ≡-refl
;F-resp≡      = λ f≈g → (λ x → f≈g {x}) , ≡-refl
}
-- (MkRel X ⊥ ⊥ → Alg) ≅ (X → One Alg)
Left1 : (ℓ ℓ' : Level) → Adjunction (Free1 ℓ ℓ') (Forget1 ℓ ℓ')
Left1 ℓ ℓ' = record
  {unit = record
    {η = λ _ → id
;commute = λ _ → ≡.refl
}
;counit = record
  {η = λ A → MkHom (λ z → z) (λ { {()}}) (λ {x} { })
;commute = λ f → ≡-refl , (λ ())
}
;zig = ≡-refl , (λ ())
;zag = ≡.refl
}

```

$\text{CoFree}^1 : (\ell : \text{Level}) \rightarrow \text{Functor} (\text{Sets } \ell) (\text{Rels } \ell \ell)$

$\text{CoFree}^1 \ell = \text{record}$

```
{F0          = λ A → MkHRel A ⊤ (λ _ → A)
;F1          = λ f → MkHom f id f
;identity     = ≐-refl , ≐-refl
;homomorphism = ≐-refl , ≐-refl
;F-resp≡     = λ f≈g → (λ x → f≈g {x}) , ≐-refl
}
```

-- (One Alg \longrightarrow X) \cong (Alg \longrightarrow MkRel X \top (λ _ → X))

$\text{Right}^1 : (\ell : \text{Level}) \rightarrow \text{Adjunction} (\text{Forget}^1 \ell \ell) (\text{CoFree}^1 \ell)$

$\text{Right}^1 \ell = \text{record}$

```
{unit = record
  {η = λ _ → MkHom id (λ _ → tt) (λ {x} {y} _ → x)
  ; commute = λ _ → ≐-refl , (λ x → ≡.refl)}
}
; counit = record {η = λ _ → id; commute = λ _ → ≡.refl}
; zig    = ≡.refl
; zag    = ≐-refl , λ {tt → ≡.refl}
}
```

-- Another cofree functor:

$\text{CoFree}^{1'} : (\ell : \text{Level}) \rightarrow \text{Functor} (\text{Sets } \ell) (\text{Rels } \ell \ell)$

$\text{CoFree}^{1'} \ell = \text{record}$

```
{F0          = λ A → MkHRel A ⊤ (λ _ → ⊤)
;F1          = λ f → MkHom f id id
;identity     = ≐-refl , ≐-refl
;homomorphism = ≐-refl , ≐-refl
;F-resp≡     = λ f≈g → (λ x → f≈g {x}) , ≐-refl
}
```

-- (One Alg \longrightarrow X) \cong (Alg \longrightarrow MkRel X \top (λ _ → ⊤))

$\text{Right}^{1'} : (\ell : \text{Level}) \rightarrow \text{Adjunction} (\text{Forget}^1 \ell \ell) (\text{CoFree}^{1'} \ell)$

$\text{Right}^{1'} \ell = \text{record}$

```
{unit = record
  {η = λ _ → MkHom id (λ _ → tt) (λ {x} {y} _ → tt)
  ; commute = λ _ → ≐-refl , (λ x → ≡.refl)}
}
; counit = record {η = λ _ → id; commute = λ _ → ≡.refl}
; zig    = ≡.refl
; zag    = ≐-refl , λ {tt → ≡.refl}
}
```

But wait, adjoints are necessarily unique, up to isomorphism, whence $\text{CoFree}^1 \cong \text{Cofree}^{1'}$. Intuitively, the relation part is a “subset” of the given carriers and when one of the carriers is a singleton then the largest relation is the universal relation which can be seen as either the first non-singleton carrier or the “always-true” relation which happens to be formalized by ignoring its arguments and going to a singleton set.

Candidate adjoints to forgetting the *second* component of a Rels

$\text{Free}^2 : (\ell : \text{Level}) \rightarrow \text{Functor} (\text{Sets } \ell) (\text{Rels } \ell \ell)$

$\text{Free}^2 \ell = \text{record}$

```
{F0          = λ A → MkHRel ⊥ A (λ ())
;F1          = λ f → MkHom id f (λ {})
;identity     = ≐-refl , ≐-refl
;homomorphism = ≐-refl , ≐-refl
}
```



```

;F-resp-≡ = λ F≈G → ≐-refl , (λ x → F≈G {x})
}
-- (MkRel ⊥ X ⊥ → Alg) ≅ (X → Two Alg)
Left2 : (ℓ : Level) → Adjunction (Free2 ℓ) (Forget2 ℓ ℓ)
Left2 ℓ = record
{unit = record
  {η = λ _ → id
   ; commute = λ _ → ≡.refl
  }
; counit = record
  {η = λ _ → MkHom (λ ()) id (λ { })
   ; commute = λ f → (λ ()) , ≐-refl
  }
; zig = (λ ()) , ≐-refl
; zag = ≡.refl
}

CoFree2 : (ℓ : Level) → Functor (Sets ℓ) (Rels ℓ ℓ)
CoFree2 ℓ = record
{F0      = λ A → MkHRel ⊤ A (λ _ _ → ⊤)
;F1      = λ f → MkHom id f id
;identity  = ≐-refl , ≐-refl
;homomorphism = ≐-refl , ≐-refl
;F-resp-≡ = λ F≈G → ≐-refl , (λ x → F≈G {x})
}
-- (Two Alg → X) ≅ (Alg → ⊤ X ⊤)
Right2 : (ℓ : Level) → Adjunction (Forget2 ℓ ℓ) (CoFree2 ℓ)
Right2 ℓ = record
{unit = record
  {η = λ _ → MkHom (λ _ → tt) id (λ _ → tt)
   ; commute = λ f → ≐-refl , ≐-refl
  }
; counit = record
  {η = λ _ → id
   ; commute = λ _ → ≡.refl
  }
; zig = ≡.refl
; zag = (λ {tt → ≡.refl}) , ≐-refl
}

```

Candidate adjoints to forgetting the *third* component of a Rels

```

Free3 : (ℓ : Level) → Functor (Twos ℓ) (Rels ℓ ℓ)
Free3 ℓ = record
{F0      = λ S → MkHRel (One S) (Two S) (λ _ _ → ⊥)
;F1      = λ f → MkHom (one f) (two f) id
;identity  = ≐-refl , ≐-refl
;homomorphism = ≐-refl , ≐-refl
;F-resp-≡ = id
} where open TwoSorted; open TwoHom
-- (MkTwo X Y → Alg without Rel) ≅ (MkRel X Y ⊥ → Alg)
Left3 : (ℓ : Level) → Adjunction (Free3 ℓ) (Forget3 ℓ ℓ)
Left3 ℓ = record
{unit = record

```

```

{η = λ A → MkTwoHom id id
; commute = λ F → ≐-refl , ≐-refl
}
; counit = record
{η = λ A → MkHom id id (λ ())
; commute = λ F → ≐-refl , ≐-refl
}
; zig = ≐-refl , ≐-refl
; zag = ≐-refl , ≐-refl
}

```

$\text{CoFree}^3 : (\ell : \text{Level}) \rightarrow \text{Functor} (\text{Twos } \ell) (\text{Rels } \ell \ell)$

```

CoFree3 ℓ = record
{F0      = λ S → MkHRel (One S) (Two S) (λ _ _ → τ)
; F1      = λ f → MkHom (one f) (two f) id
; identity = ≐-refl , ≐-refl
; homomorphism = ≐-refl , ≐-refl
; F-resp≡ = id
} where open TwoSorted; open TwoHom
-- (Alg without Rel → MkTwo X Y) ≅ (Alg → MkRel X Y τ)

```

$\text{Right}^3 : (\ell : \text{Level}) \rightarrow \text{Adjunction} (\text{Forget}^3 \ell \ell) (\text{CoFree}^3 \ell)$

```

Right3 ℓ = record
{unit = record
{η = λ A → MkHom id id (λ _ → tt)
; commute = λ F → ≐-refl , ≐-refl
}
; counit = record
{η = λ A → MkTwoHom id id
; commute = λ F → ≐-refl , ≐-refl
}
; zig = ≐-refl , ≐-refl
; zag = ≐-refl , ≐-refl
}

```

$\text{CoFree}^{3'} : (\ell : \text{Level}) \rightarrow \text{Functor} (\text{Twos } \ell) (\text{Rels } \ell \ell)$

```

CoFree3' ℓ = record
{F0      = λ S → MkHRel (One S) (Two S) (λ _ _ → One S × Two S)
; F1      = λ F → MkHom (one F) (two F) (one F ×1 two F)
; identity = ≐-refl , ≐-refl
; homomorphism = ≐-refl , ≐-refl
; F-resp≡ = id
} where open TwoSorted; open TwoHom
-- (Alg without Rel → MkTwo X Y) ≅ (Alg → MkRel X Y X×Y)

```

$\text{Right}^{3'} : (\ell : \text{Level}) \rightarrow \text{Adjunction} (\text{Forget}^3 \ell \ell) (\text{CoFree}^{3'} \ell)$

```

Right3' ℓ = record
{unit = record
{η = λ A → MkHom id id (λ {x} {y} x~y → x , y)
; commute = λ F → ≐-refl , ≐-refl
}
; counit = record
{η = λ A → MkTwoHom id id
; commute = λ F → ≐-refl , ≐-refl
}
; zig = ≐-refl , ≐-refl
; zag = ≐-refl , ≐-refl
}

```

But wait, adjoints are necessarily unique, up to isomorphism, whence $\text{CoFree}^3 \cong \text{CoFree}^{3'}$. Intuitively, the relation part is a “subset” of the given carriers and so the largest relation is the universal relation which can be seen as the product of the carriers or the “always-true” relation which happens to be formalized by ignoring its arguments and going to a singleton set.

7.4 ???

It remains to port over results such as Merge, Dup, and Choice from Twos to Rels.

Also to consider: sets with an equivalence relation; whence propositional equality.

The category of sets contains products and so **TwoSorted** algebras can be represented there and, moreover, this is adjoint to duplicating a type to obtain a **TwoSorted** algebra.

-- The category of Sets has products and so the **TwoSorted** type can be reified there.

Merge : (ℓ : Level) → Functor (Twos ℓ) (Sets ℓ)

Merge ℓ = **record**

```
{F0          = λ S → One S × Two S
;F1          = λ F → one F ×1 two F
;identity     = ≡.refl
;homomorphism = ≡.refl
;F-resp≡     = λ {(F≈1G , F≈2G) {x , y} → ≡.cong2 _ , _ (F≈1G x) (F≈2G y)}
}
```

-- Every set gives rise to its square as a **TwoSorted** type.

Dup : (ℓ : Level) → Functor (Sets ℓ) (Twos ℓ)

Dup ℓ = **record**

```
{F0          = λ A → MkTwo A A
;F1          = λ f → MkHom f f
;identity     = ≡-refl , ≡-refl
;homomorphism = ≡-refl , ≡-refl
;F-resp≡     = λ F≈G → diag (λ _ → F≈G)
}
```

Then the proof that these two form the desired adjunction

Right₂ : (ℓ : Level) → Adjunction (Dup ℓ) (Merge ℓ)

Right₂ ℓ = **record**

```
{unit        = record {η = λ _ → diag; commute = λ _ → ≡.refl}
; counit     = record {η = λ _ → MkHom proj1 proj2; commute = λ _ → ≡-refl , ≡-refl}
; zig       = ≡-refl , ≡-refl
; zag       = ≡.refl
}
```

The category of sets admits sums and so an alternative is to represent a **TwoSorted** algebra as a sum, and moreover this is adjoint to the aforementioned duplication functor.

Choice : (ℓ : Level) → Functor (Twos ℓ) (Sets ℓ)

Choice ℓ = **record**

```
{F0          = λ S → One S ⊔ Two S
;F1          = λ F → one F ⊔1 two F
;identity     = ⊔-id $i
;homomorphism = λ { {x = x} → ⊔-o x }
;F-resp≡     = λ F≈G {x} → uncurry ⊔-cong F≈G x
}
```

Left₂ : (ℓ : Level) → Adjunction (Choice ℓ) (Dup ℓ)

Left₂ ℓ = **record**

```

{unit    = record {η = λ _ → MkHom inj1 inj2; commute = λ _ → ≐-refl , ≐-refl}
; counit = record {η = λ _ → from⊖; commute = λ _ {x} → (≡.sym ∘ from⊖-nat) x}
; zig    = λ { {- } } {x} → from⊖-preInverse x}
; zag    = ≐-refl , ≐-refl
}

```

8 Pointed Algebras: Nullable Types

We consider the theory of *pointed algebras* which consist of a type along with an elected value of that type.¹ Software engineers encounter such scenarios all the time in the case of an object-type and a default value of a “null”, or undefined, object. In the more explicit setting of pure functional programming, this concept arises in the form of **Maybe**, or **Option** types.

Some programming languages, such as **C#** for example, provide a **default** keyword to access a default value of a given data type.

[MA: insert: Haskell’s typeclass analogue of **default**?]

[MA: Perhaps discuss “types as values” and the subtle issue of how pointed algebras are completely different than classes in an imperative setting.]

module Structures.Pointed **where**

```

open import Level renaming (suc to lsuc; zero to lzero)
open import Categories.Category using (Category; module Category)
open import Categories.Functor using (Functor)
open import Categories.Adjunction using (Adjunction)
open import Categories.NaturalTransformation using (NaturalTransformation)
open import Categories.Agda using (Sets)
open import Function using (id; _ ∘ _)
open import Data.Maybe using (Maybe; just; nothing; maybe; maybe')
open import Forget
open import Data.Empty
open import Relation.Nullary
open import EqualityCombinators

```

8.1 Definition

As mentioned before, a **Pointed algebra** is a type, which we will refer to by **Carrier**, along with a value, or **point**, of that type.

```

record Pointed {a} : Set (lsuc a) where
  constructor MkPointed
  field
    Carrier : Set a
    point   : Carrier
open Pointed

```

Unsurprisingly, a “structure preserving operation” on such structures is a function between the underlying carriers that takes the source’s point to the target’s point.

¹Note that this definition is phrased as a “dependent product”!

```

record Hom {ℓ} (X Y : Pointed {ℓ}) : Set ℓ where
  constructor MkHom
  field
    mor          : Carrier X → Carrier Y
    preservation : mor (point X) ≡ point Y
open Hom

```

8.2 Category and Forgetful Functors

Since there is only one type, or sort, involved in the definition, we may hazard these structures as “one sorted algebras”:

```

oneSortedAlg : ∀ {ℓ} → OneSortedAlg ℓ
oneSortedAlg = record
  { Alg      = Pointed
  ; Carrier  = Carrier
  ; Hom      = Hom
  ; mor      = mor
  ; comp     = λ F G → MkHom (mor F ∘ mor G) (≡.cong (mor F) (preservation G) (≡≡) preservation F)
  ; comp-is-∘ = ≡-refl
  ; Id       = MkHom id ≡.refl
  ; Id-is-id = ≡-refl
  }

```

From which we immediately obtain a category and a forgetful functor.

```

Pointeds : (ℓ : Level) → Category (Isuc ℓ) ℓ ℓ
Pointeds ℓ = oneSortedCategory ℓ oneSortedAlg
Forget : (ℓ : Level) → Functor (Pointeds ℓ) (Sets ℓ)
Forget ℓ = mkForgetful ℓ oneSortedAlg

```

The naming `Pointeds` is to be consistent with the category theory library we are using, which names the category of sets and functions by `Sets`. That is, the category name is the objects’ name suffixed with an ‘s’.

Of-course, as hinted in the introduction, this structure —as are many— is defined in a dependent fashion and so we have another forgetful functor:

```

open import Data.Product
ForgetD : (ℓ : Level) → Functor (Pointeds ℓ) (Sets ℓ)
ForgetD ℓ = record { F0 = λ P → Σ (Carrier P) (λ x → x ≡ point P)
  ; F1 = λ {P} {Q} F → λ {(val , val≡ptP) → mor F val , (≡.cong (mor F) val≡ptP (≡≡) preservation F)}
  ; identity = λ {P} → λ {(val , val≡ptP) → ≡.cong (λ x → val , x) (≡.proof-irrelevance _ _)}
  ; homomorphism = λ {P} {Q} {R} {F} {G} → λ {(val , val≡ptP) → ≡.cong (λ x → mor G (mor F val) , x) (≡.proof-irrelevance _ _)}
  ; F-resp≡ = λ {P} {Q} {F} {G} F≈G → λ {(val , val≡ptP) → {!≡.cong2 _ _ (F≈G val) ?!}}
  }

```

That is, we “only remember the point”.

[MA: insert:] An adjoint to this functor? **[]**

8.3 A Free Construction

As discussed earlier, the prime example of pointed algebras are the optional types, and this claim can be realised as a functor:

```

Free : (ℓ : Level) → Functor (Sets ℓ) (Pointeds ℓ)
Free ℓ = record
  { F0      = λ A → MkPointed (Maybe A) nothing
  ; F1      = λ f → MkHom (maybe (just ∘ f) nothing) ≡.refl
  ; identity  = maybe ≡-refl ≡.refl
  ; homomorphism = maybe ≡-refl ≡.refl
  ; F-resp-≡ = λ F≡G → maybe (o-resp-≡ (≡-refl {x = just})) (λ x → F≡G {x})) ≡.refl
  }

```

Which is indeed deserving of its name:

```

MaybeLeft : (ℓ : Level) → Adjunction (Free ℓ) (Forget ℓ)
MaybeLeft ℓ = record
  { unit      = record { η = λ _ → just; commute = λ _ → ≡.refl }
  ; counit    = record
    { η      = λ X → MkHom (maybe id (point X)) ≡.refl
    ; commute = maybe ≡-refl ∘ ≡.sym ∘ preservation
    }
  ; zig       = maybe ≡-refl ≡.refl
  ; zag       = ≡.refl
  }

```

[MA:] *Develop Maybe explicitly so we can “see” how the utility maybe “pops up naturally”.* **[1]**

While there is a “least” pointed object for any given set, there is, in-general, no “largest” pointed object corresponding to any given set. That is, there is no co-free functor.

```

NoRight : {ℓ : Level} → (CoFree : Functor (Sets ℓ) (Pointeds ℓ)) → ¬ (Adjunction (Forget ℓ) CoFree)
NoRight (record { F0 = f }) Adjunct = lower (η (counit Adjunct) (Lift ⊥) (point (f (Lift ⊥))))
  where open Adjunction
  open NaturalTransformation

```

9 SetoidSetoid

module SetoidSetoid **where**

```

open import Level renaming (zero to lzero; suc to lsuc; _⊔_ to _⊔_ ) hiding (lift)
open import Relation.Binary using (Setoid)
open import DataProperties using (T; tt)
open import SetoidEquiv

```

Setoid of setoids SSetoid, and “setoid” of equality proofs.

```

SSetoid : (ℓ o : Level) → Setoid (lsuc o ⊔ lsuc ℓ) (o ⊔ ℓ)
SSetoid ℓ o = record
  { Carrier = Setoid ℓ o
  ; _≈_      = _≅_
  ; isEquivalence = record { refl = ≅-refl; sym = ≅-sym; trans = ≅-trans } }

```

Given two elements of a given Setoid A, define a Setoid of equivalences of those elements. We consider all such equivalences to be equivalent. In other words, for $e_1 e_2 : \text{Setoid.Carrier } A$, then $e_1 \approx_s e_2$, as a type, is contractible.

```

_≈S_ : ∀ {a ℓa} {A : Setoid a ℓa} → (e1 e2 : Setoid.Carrier A) → Setoid ℓa ℓa
_≈S_ {A = A} e1 e2 = let open Setoid A renaming (_≈_ to _≈s_ ) in
  record { Carrier = e1 ≈s e2; _≈_ = λ _ _ → T
  ; isEquivalence = record { refl = tt; sym = λ _ → tt; trans = λ _ _ → tt } }

```

10 Some

```

module Some where
open import Level renaming (zero to lzero; suc to lsuc) hiding (lift)
open import Relation.Binary using (Setoid; lsEquivalence; Rel;
  Reflexive; Symmetric; Transitive)
open import Function.Equality using ( $\Pi$ ;  $\_ \longrightarrow \_$ ; id;  $\_ \circ \_$ ;  $\_ \langle \$ \rangle \_$ )
open import Function using ( $\_ \$ \_$ ) renaming (id to id0;  $\_ \circ \_$  to  $\_ \odot \_$ )
open import Data.List using (List; [];  $\_ ++ \_$ ;  $\_ :: \_$ ; map)
open import Data.Product using ( $\exists$ )
open import Data.Nat using ( $\mathbb{N}$ ; zero; suc)
open import EqualityCombinators
open import DataProperties
open import SetoidEquiv
open import TypeEquiv using (swap+)
open import SetoidSetoid
open import Relation.Binary.Sum
open import Relation.Binary.PropositionalEquality using (inspect)

```

[WK: Goal?]

10.1 Some₀

Setoid based variant of Any.

Quite a bit of this is directly inspired by Data.List.Any and Data.List.Any.Properties.

[WK: $A \longrightarrow SSetoid _ _$ is a pretty strong assumption. Logical equivalence does not ask for the two morphisms back and forth to be inverse.]

```

module  $\_ \{a \ell a\} \{A : Setoid a \ell a\} (P : A \longrightarrow SSetoid \ell a \ell a)$  where
  open Setoid A
  private P0 =  $\lambda e \rightarrow Setoid.Carrier (\Pi. \_ \langle \$ \rangle \_ P e)$ 
  data Some0 : List Carrier  $\rightarrow Set (a \sqcup \ell a)$  where
    here :  $\{x : Carrier\} \{xs : List Carrier\} (px : P_0 x) \rightarrow Some_0 (x :: xs)$ 
    there :  $\{x : Carrier\} \{xs : List Carrier\} (pxs : Some_0 xs) \rightarrow Some_0 (x :: xs)$ 

```

Inhabitants of Some₀ really are just locations: Some₀ P xs $\cong \sum i : Fin (length xs) \bullet P (x ! i)$. Thus one possibility is to go with natural numbers directly, and entirely ignore the proof contained in a Some₀ P xs.

```

to $\mathbb{N}$  :  $\{xs : List Carrier\} \rightarrow Some_0 xs \rightarrow \mathbb{N}$ 
to $\mathbb{N}$  (here  $\_$ ) = 0
to $\mathbb{N}$  (there pf) = suc (to $\mathbb{N}$  pf)
 $\_ \sim_S \_$  :  $\{xs : List Carrier\} \rightarrow Some_0 xs \rightarrow Some_0 xs \rightarrow Set$ 
s1  $\sim_S$  s2 = to $\mathbb{N}$  s1  $\equiv$  to $\mathbb{N}$  s2

```

Instead, we choose a more direct approach: $_ \approx _$. This is an extremely strong relation: two proofs, of different properties of elements of different lists are considered related when the “witness” for the property is in the same location in both lists.

```

module  $\_ \{a \ell a\} \{A : Setoid a \ell a\} \{P : A \longrightarrow SSetoid \ell a \ell a\} \{Q : A \longrightarrow SSetoid \ell a \ell a\}$  where
  open Setoid A

```

```

private P0 = λ e → Setoid.Carrier (Π. _⟨$⟩_ P e)
private Q0 = λ e → Setoid.Carrier (Π. _⟨$⟩_ Q e)
infix 3 _≈_
data _≈_ : {xs ys : List Carrier} (pf : Some0 P xs) (pf' : Some0 Q ys) → Set ℓa where
  hereEq : {xs ys : List Carrier} {x y : Carrier} (px : P0 x) (qy : Q0 y)
    → _≈_ (here {x = x} {xs} px) (here {x = y} {ys} qy)
  thereEq : {xs ys : List Carrier} {x y : Carrier} {pxs : Some0 P xs} {qys : Some0 Q ys}
    → _≈_ pxs qys → _≈_ (there {x = x} pxs) (there {x = y} qys)
module _ {a ℓa} {A : Setoid a ℓa} {P : A → SSetoid ℓa ℓa} where
  open Setoid A
  ≈-refl : {xs : List Carrier} {p : Some0 P xs} → p ≈ p
  ≈-refl {p = here px} = hereEq px px
  ≈-refl {p = there p} = thereEq ≈-refl
module _ {a ℓa} {A : Setoid a ℓa} {P : A → SSetoid ℓa ℓa} {Q : A → SSetoid ℓa ℓa} where
  open Setoid A
  ≈-sym : {xs : List Carrier} {p : Some0 P xs} {q : Some0 Q xs} → p ≈ q → q ≈ p
  ≈-sym (hereEq px py) = hereEq py px
  ≈-sym (thereEq eq) = thereEq (≈-sym eq)
module _ {a ℓa} {A : Setoid a ℓa} {P : A → SSetoid ℓa ℓa} {Q : A → SSetoid ℓa ℓa} {R : A → SSetoid ℓa ℓa} where
  open Setoid A
  ≈-trans : {xs : List Carrier} {p : Some0 P xs} {q : Some0 Q xs} {r : Some0 R xs}
    → p ≈ q → q ≈ r → p ≈ r
  ≈-trans (hereEq px py) (hereEq .py pz) = hereEq px pz
  ≈-trans (thereEq e) (thereEq f) = thereEq (≈-trans e f)
module _ {a ℓa} {A : Setoid a ℓa} (P : A → SSetoid ℓa ℓa) where
  open Setoid A
  private P0 = λ e → Setoid.Carrier (Π. _⟨$⟩_ P e)
  Some : List Carrier → Setoid (ℓa ⊔ a) ℓa
  Some xs = record
    { Carrier      = Some0 P xs
    ; _≈_          = _≈_
    ; isEquivalence = record { refl = ≈-refl; sym = ≈-sym; trans = ≈-trans }
    -- record { isEquivalence ≡ isEquivalence }
    }
  ⇒→Some : {a ℓa : Level} {A : Setoid a ℓa} {P : A → SSetoid ℓa ℓa}
    {xs ys : List (Setoid.Carrier A)} → xs ≡ ys → Some P xs ≡ Some P ys
  ⇒→Some {A = A} ≡.refl = ≡-refl

```

10.2 Membership module

```

module Membership {a ℓ} (S : Setoid a ℓ) where
  open Setoid S renaming (trans to _⟨≈≈⟩_)
  infix 4 _∈0 _ ∈_

```

setoid≈ x is actually a mapping from S to SSetoid _; it maps elements y of Carrier S to the setoid of "x ≈_s y".

```

setoid≈ : Carrier → S → SSetoid ℓ ℓ
setoid≈ x = record
  { _⟨$⟩_ = λ (y : Carrier) → _≈S_ {A = S} x y
  ; cong = λ i≈j → record
    { to = record { _⟨$⟩_ = λ x≈i → x≈i ⟨≈≈⟩ i≈j; cong = λ _ → tt }
    ; from = record { _⟨$⟩_ = λ x≈j → x≈j ⟨≈≈⟩ sym i≈j; cong = λ _ → tt }
    }

```



```

;inverse-of = record
  {left-inverse-of = λ _ → tt
   ;right-inverse-of = λ _ → tt
  }
}

_ ∈ _ : Carrier → List Carrier → Setoid (a ⊔ ℓ) ℓ
x ∈ xs = Some (setoid≈ x) xs

_ ∈0 _ : Carrier → List Carrier → Set (ℓ ⊔ a)
x ∈0 xs = Setoid.Carrier (x ∈ xs)

BagEq : (xs ys : List Carrier) → Set (ℓ ⊔ a)
BagEq xs ys = {x : Carrier} → (x ∈ xs) ≅ (x ∈ ys)

```

10.3 Parallel Composition

To avoid absurd patterns that we do not use, when using `_⊔-Rel_`, we make this: As such, we introduce the parallel composition of heterogeneous relations.

```

data _||_ {a1 b1 c1 a2 b2 c2 : Level}
  {A1 : Set a1} {B1 : Set b1} {_~1_ : A1 → B1 → Set c1}
  {A2 : Set a2} {B2 : Set b2} {_~2_ : A2 → B2 → Set c2}
  : A1 ⊔ A2 → B1 ⊔ B2 → Set (a1 ⊔ b1 ⊔ c1 ⊔ a2 ⊔ b2 ⊔ c2) where
  left : {x : A1} {y : B1} (x~1y : x ~1 y) → (_~1_ || _~2_ ) (inj1 x) (inj1 y)
  right : {x : A2} {y : B2} (x~2y : x ~2 y) → (_~1_ || _~2_ ) (inj2 x) (inj2 y)
  -- Non-working “eliminator” for this type.
[ _||_ ] : {a1 b1 c1 a2 b2 c2 ℓ : Level}
  {A1 : Set a1} {B1 : Set b1} {_~1_ : A1 → B1 → Set c1}
  {A2 : Set a2} {B2 : Set b2} {_~2_ : A2 → B2 → Set c2}
  →
    {Z : {a : A1 ⊔ A2} {b : B1 ⊔ B2} → (_~1_ || _~2_ ) a b → Set ℓ}
    (F : {a : A1} {b : B1} (a~b : a ~1 b) → Z (left a~b))
    (G : {a : A2} {b : B2} (a~b : a ~2 b) → Z (right a~b))
  →
    {x : A1 ⊔ A2} {y : B1 ⊔ B2}
    → (x||y : (_~1_ || _~2_ ) x y) → Z x||y
[ F || G ] (left x~y) = F x~y
[ F || G ] (right x~y) = G x~y
  -- If the argument relations are symmetric then so is their parallel composition.
||-sym : {a a' c c' : Level} {A : Set a} {_~_ : A → A → Set c}
  {A' : Set a'} {_~'_ : A' → A' → Set c'}
  (sym1 : {x y : A} → x ~ y → y ~ x) (sym2 : {x y : A'} → x ~' y → y ~' x)
  {x y : A ⊔ A'}
  →
    (_~_ || _~'_ ) x y → (_~_ || _~'_ ) y x
||-sym sym1 sym2 (left x~y) = left (sym1 x~y)
||-sym sym1 sym2 (right x~y) = right (sym2 x~y)
  --
  -- ought to be just: [ left ∘ sym1 || right ∘ sym2 ]
  --
  -- Instead, I can use, with much distasteful yellow,
  -- ||-sym sym1 sym2 = [ (λ pf → left (sym1 pf)) || (λ pf → right (sym2 pf)) ]
infix 999 _⊔_
_⊔_ : {i1 i2 k1 k2 : Level} → Setoid i1 k1 → Setoid i2 k2 → Setoid (i1 ⊔ i2) (i1 ⊔ i2 ⊔ k1 ⊔ k2)
A ⊔ B = record

```

```

{Carrier = A0 ⊔ B0
; _ ≈ _ = ≈1 || ≈2
; isEquivalence = record
  { refl = λ { {inj1 x} → left refl1; {inj2 x} → right refl2 }
  ; sym = λ { (left eq) → left (sym1 eq); (right eq) → right (sym2 eq) }
    -- ought to be writable as [ left ∘ sym1 || right ∘ sym2 ]
  ; trans = λ { (left eq) (left eqq) → left (trans1 eq eqq)
    ; (right eq) (right eqq) → right (trans2 eq eqq)
    }
  }
}
where
  open Setoid A renaming (Carrier to A0; _ ≈ _ to ≈1; refl to refl1; sym to sym1; trans to trans1)
  open Setoid B renaming (Carrier to B0; _ ≈ _ to ≈2; refl to refl2; sym to sym2; trans to trans2)

```

10.4 ⊔-comm

```

⊔-comm : {a b aℓ bℓ : Level} {A : Setoid a aℓ} {B : Setoid b bℓ} → A ⊔ B ≅ B ⊔ A
⊔-comm {A = A} {B} = record
  { to = record { _ ($) _ = swap+; cong = swap-on-|| }
  ; from = record { _ ($) _ = swap+; cong = swap-on-||' }
  ; inverse-of = record { left-inverse-of = swap2≈||≈id; right-inverse-of = swap2≈||≈id' }
  }
where
  open Setoid A renaming (Carrier to A0; _ ≈ _ to ≈1; refl to refl1)
  open Setoid B renaming (Carrier to B0; _ ≈ _ to ≈2; refl to refl2)

  swap-on-|| : {i j : A0 ⊔ B0} → (≈1 || ≈2) i j → (≈2 || ≈1) (swap+ i) (swap+ j)
  swap-on-|| (left x~1y) = right x~1y
  swap-on-|| (right x~2y) = left x~2y

  swap2≈||≈id : (z : A0 ⊔ B0) → (≈1 || ≈2) (swap+ (swap+ z)) z
  swap2≈||≈id (inj1 _) = left refl1
  swap2≈||≈id (inj2 _) = right refl2

  {-Tried to obtain the following via ||-sym ... -}
  swap-on-||' : {i j : B0 ⊔ A0} → (≈2 || ≈1) i j → (≈1 || ≈2) (swap+ i) (swap+ j)
  swap-on-||' (left x~y) = right x~y
  swap-on-||' (right x~y) = left x~y

  swap2≈||≈id' : (z : B0 ⊔ A0) → (≈2 || ≈1) (swap+ (swap+ z)) z
  swap2≈||≈id' (inj1 _) = left refl2
  swap2≈||≈id' (inj2 _) = right refl1

```

10.5 ++≅ : ... → (Some P xs ⊔ Some P ys) ≅ Some P (xs ++ ys)

```

module _ {a ℓa : Level} {A : Setoid a ℓa} {P : A → SSetoid ℓa ℓa} where
  ++≅ : {xs ys : List (Setoid.Carrier A)} → (Some P xs ⊔ Some P ys) ≅ Some P (xs ++ ys)
  ++≅ {xs} {ys} = record
    { to = record { _ ($) _ = ⊔→++; cong = ⊔→++-cong }
    ; from = record { _ ($) _ = ++→⊔; cong = new-cong xs }
    ; inverse-of = record
      { left-inverse-of = lefty xs
      ; right-inverse-of = righty xs
      }
    }

```

```

}
where
  open Setoid A
  _~_ = _~S_ P
  _~_ = _~_ ; ~-refl = ~-refl {P = P}
  -- “ealier”
   $\mapsto^l : \forall \{ws\ zs\} \rightarrow \text{Some}_0 P \ ws \rightarrow \text{Some}_0 P (ws + zs)$ 
   $\mapsto^l (\text{here } p) = \text{here } p$ 
   $\mapsto^l (\text{there } p) = \text{there } (\mapsto^l p)$ 

```

The following absurd patterns are what led me to make a new type for equalities.

```

yo : {xs : List Carrier} {x y : Some0 P xs} → x ~ y →  $\mapsto^l x \sim \mapsto^l y$ 
yo {x = here px} {here px1} Relation.Binary.PropositionalEquality.refl =  $\equiv$ .refl
yo {x = here px} {there y} ()
yo {x = there x1} {here px} ()
yo {x = there x1} {there y} pf =  $\equiv$ .cong suc (yo {!!})

```

```

yo : {xs : List Carrier} {x y : Some0 P xs} → x ~ y →  $\mapsto^l x \sim \mapsto^l y$ 
yo (hereEq px py) = hereEq px py
yo (thereEq pf) = thereEq (yo pf)
  -- “later”
 $\mapsto^r : \forall \{xs\ ys\} \rightarrow \text{Some}_0 P \ ys \rightarrow \text{Some}_0 P (xs + ys)$ 
 $\mapsto^r [] p = p$ 
 $\mapsto^r (x :: xs) p = \text{there } (\mapsto^r xs p)$ 
oy : (xs : List Carrier) {x y : Some0 P ys} → x ~ y →  $\mapsto^r xs \times \sim \mapsto^r xs y$ 
oy [] pf = pf
oy (x :: xs) pf = thereEq (oy xs pf)

```

-- Some₀ is $++\rightarrow\uplus$ -homomorphic, in the second argument.

```

 $\mapsto^{++} : \forall \{zs\ ws\} \rightarrow (\text{Some}_0 P \ zs \uplus \text{Some}_0 P \ ws) \rightarrow \text{Some}_0 P (zs + ws)$ 
 $\mapsto^{++} (\text{inj}_1 x) = \mapsto^l x$ 
 $\mapsto^{++} \{zs\} (\text{inj}_2 y) = \mapsto^r zs y$ 
 $++\rightarrow\uplus : \forall \{xs\ ys\} \rightarrow \text{Some}_0 P (xs + ys) \rightarrow \text{Some}_0 P \ xs \uplus \text{Some}_0 P \ ys$ 
 $++\rightarrow\uplus [] p = \text{inj}_2 p$ 
 $++\rightarrow\uplus (x :: l) (\text{here } p) = \text{inj}_1 (\text{here } p)$ 
 $++\rightarrow\uplus (x :: l) (\text{there } p) = (\text{there } \uplus_1 \text{ id}_0) (++\rightarrow\uplus l p)$ 

```

-- all of the following may need to change

```

 $\mapsto^{++}\text{-cong} : \{a\ b : \text{Some}_0 P \ xs \uplus \text{Some}_0 P \ ys\} \rightarrow (\_ \sim \_ \parallel \_ \sim \_) a\ b \rightarrow \mapsto^{++} a \sim \mapsto^{++} b$ 
 $\mapsto^{++}\text{-cong} (\text{left } x_1 \sim x_2) = \text{yo } x_1 \sim x_2$ 
 $\mapsto^{++}\text{-cong} (\text{right } y_1 \sim y_2) = \text{oy } xs\ y_1 \sim y_2$ 
 $\sim \parallel \sim\text{-cong} : \{xs\ ys\ us : \text{List Carrier}\}$ 
  → (F : Some0 P xs → Some0 P us) (F-cong : {p q : Some0 P xs} → p ~ q → F p ~ F q)
  → (G : Some0 P ys → Some0 P vs) (G-cong : {p q : Some0 P ys} → p ~ q → G p ~ G q)
  → {pf pf' : Some0 P xs uplus Some0 P ys}
  → ( $\_ \sim \_ \parallel \_ \sim \_$ ) pf pf' → ( $\_ \sim \_ \parallel \_ \sim \_$ ) ((F  $\uplus_1$  G) pf) ((F  $\uplus_1$  G) pf')
 $\sim \parallel \sim\text{-cong } F\ F\text{-cong } G\ G\text{-cong} (\text{left } x_1 \sim x_2) = \text{left } (F\text{-cong } x_1 \sim x_2)$ 
 $\sim \parallel \sim\text{-cong } F\ F\text{-cong } G\ G\text{-cong} (\text{right } x_1 \sim x_2) = \text{right } (G\text{-cong } x_1 \sim x_2)$ 
new-cong : (xs : List Carrier) {i j : Some0 P (xs + ys)} → i ~ j → ( $\_ \sim \_ \parallel \_ \sim \_$ ) (++ $\rightarrow\uplus$  xs i) (++ $\rightarrow\uplus$  xs j)
new-cong [] pf = right pf
new-cong (x :: xs) (hereEq px py) = left (hereEq px py)
new-cong (x :: xs) (thereEq pf) =  $\sim \parallel \sim\text{-cong}$  there thereEq id0 id0 (new-cong xs pf)
lefty : (xs {ys} : List Carrier) (p : Some0 P xs uplus Some0 P ys) → ( $\_ \sim \_ \parallel \_ \sim \_$ ) (++ $\rightarrow\uplus$  xs ( $\mapsto^{++}$  p)) p
lefty [] (inj1 ())

```

```

lefty [] (inj2 p) = right ≈-refl
lefty (x :: xs) (inj1 (here px)) = left ~-refl
lefty (x :: xs) {ys} (inj1 (there p)) with ++→⊕ xs {ys} (⊕→++ (inj1 p)) | lefty xs {ys} (inj1 p)
... | inj1 _ | (left x~1y) = left (thereEq x~1y)
... | inj2 _ | ()
lefty (z :: zs) {ws} (inj2 p) with ++→⊕ zs {ws} (⊕→++ {zs} (inj2 p)) | lefty zs (inj2 p)
... | inj1 x | ()
... | inj2 y | (right x~2y) = right x~2y
righty : (zs {ws} : List Carrier) (p : Some0 P (zs + ws)) → (⊕→++ (++→⊕ zs p)) ~ p
righty [] {ws} p = ~-refl
righty (x :: zs) {ws} (here px) = ~-refl
righty (x :: zs) {ws} (there p) with ++→⊕ zs p | righty zs p
... | inj1 _ | res = thereEq res
... | inj2 _ | res = thereEq res

```

10.6 Bottom as a setoid

```

⊥⊥ : ∀ {a ℓa} → Setoid a ℓa
⊥⊥ {a} {ℓa} = record
  {Carrier = ⊥
   ;_≈_ = λ _ _ → ⊤
   ;isEquivalence = record {refl = tt; sym = λ _ → tt; trans = λ _ _ → tt}
  }

```

```

module _ {a ℓa : Level} {A : Setoid a ℓa} {P : A → SSetoid ℓa ℓa} where
  ⊥≅Some[] : ⊥⊥ {a} {ℓa} ≅ Some P []
  ⊥≅Some[] = record
    {to = record {_⟨$⟩_ = λ {} → (); cong = λ {} {} → {} }
     ;from = record {_⟨$⟩_ = λ {} → (); cong = λ {} {} → {} }
     ;inverse-of = record {left-inverse-of = λ _ → tt; right-inverse-of = λ {} → {} }
    }

```

10.7 $\text{map}^\approx : \dots \rightarrow \text{Some } (P \circ f) \text{ xs} \cong \text{Some } P (\text{map } (_ \langle \$ \rangle _) f) \text{ xs}$

```

map≈ : ∀ {a ℓa} {A B : Setoid a ℓa} {P : B → SSetoid ℓa ℓa} {f : A → B} {xs : List (Setoid.Carrier A)} →
  Some (P ∘ f) xs ≅ Some P (map ( _ ⟨ $ ⟩ _ ) f) xs
map≈ {A = A} {B} {P} {f} = record
  {to = record {_⟨$⟩_ = map+; cong = map+-cong}
   ;from = record {_⟨$⟩_ = map-; cong = map--cong}
   ;inverse-of = record {left-inverse-of = map- ∘ map+; right-inverse-of = map+ ∘ map-}
  }
where
  g = _⟨$⟩_ f
  A0 = Setoid.Carrier A
  _~_ = _≈_ {P = P}
  map+ : {xs : List A0} → Some0 (P ∘ f) xs → Some0 P (map g xs)
  map+ (here p) = here p
  map+ (there p) = there $ map+ p
  map- : {xs : List A0} → Some0 P (map g xs) → Some0 (P ∘ f) xs
  map- {} ()
  map- {x :: xs} (here p) = here p

```

```

map⁻ {x :: xs} (there p) = there (map⁻ {xs = xs} p)
map⁺ ∘ map⁻ : {xs : List A₀} → (p : Some₀ P (map g xs)) → map⁺ (map⁻ p) ~ p
map⁺ ∘ map⁻ {[]} ()
map⁺ ∘ map⁻ {x :: xs} (here p) = hereEq p p
map⁺ ∘ map⁻ {x :: xs} (there p) = thereEq (map⁺ ∘ map⁻ p)
map⁻ ∘ map⁺ : {xs : List A₀} → (p : Some₀ (P ∘ f) xs)
→ let _~₂_ = _≈_ {P = P ∘ f} in map⁻ (map⁺ p) ~₂ p
map⁻ ∘ map⁺ {[]} ()
map⁻ ∘ map⁺ {x :: xs} (here p) = hereEq p p
map⁻ ∘ map⁺ {x :: xs} (there p) = thereEq (map⁻ ∘ map⁺ p)
map⁺-cong : {ys : List A₀} {i j : Some₀ (P ∘ f) ys} → _≈_ {P = P ∘ f} i j → map⁺ i ~ map⁺ j
map⁺-cong (hereEq px py) = hereEq px py
map⁺-cong (thereEq i~j) = thereEq (map⁺-cong i~j)
map⁻-cong : {ys : List A₀} {i j : Some₀ P (map g ys)} → i ~ j → _≈_ {P = P ∘ f} (map⁻ i) (map⁻ j)
map⁻-cong {[]} ()
map⁻-cong {x :: ys} (hereEq px py) = hereEq px py
map⁻-cong {x :: ys} (thereEq i~j) = thereEq (map⁻-cong i~j)

```

```
module FindLose {a ℓa : Level} {A : Setoid a ℓa} (P : A → SSetoid ℓa ℓa) where
```

open Membership A

open Setoid A

open Π

open 112

private

$$P_0 = \lambda e \rightarrow \text{Setoid.Carrier } (\Pi_ \langle \$ \rangle_ P e)$$
$$\text{Support} = \lambda y s \rightarrow \sum y : \text{Carrier} \bullet y \in_0 y s \times P_0 y$$
$$\text{find} : \{\text{ys} : \text{List Carrier}\} \rightarrow \text{Some}_0 P \text{ ys} \rightarrow \text{Support ys}$$
$$\text{find } \{y :: ys\} \text{ (here } p) = y, \text{ here refl, } p$$
$$\text{find } \{y :: ys\} \text{ (there } p) = \text{let } (a, a \in ys, Pa) = \text{find } p$$

$$\text{in } a, \text{there } a \in ys, Pa$$
$$\text{lose} : \{ys : \text{List Carrier}\} \rightarrow \text{Support } ys \rightarrow \text{Some}_0 P \text{ } ys$$
$$\text{lose}(y, \text{here } p_y, P_y) = \text{here}(_\cong_.\text{to}(\Pi.\text{cong } P \text{ } p_y) \Pi.\langle \$ \rangle P_y)$$
$$\text{lose}(y, \text{there } y \in y_s, P_y) = \text{there}(\text{lose}(y, y \in y_s, P_y))$$

-- “If an element of `ys` has a property `P`, then some element of `ys` has property `P`”

-- cf copy below

$$\text{Some-Intro} : \{y : \text{Carrier}\} \{ys : \text{List Carrier}\}$$
$$\rightarrow y \in_0 ys \rightarrow P_0 y \rightarrow \text{Some}_0 P ys$$

Some-Intro $\{y\} \ y \in y_s \ Qy = \text{lose}(y, y \in y_s, Qy)$

$$\text{bag-as} \Rightarrow : \{xs\ ys : \text{List Carrier}\} \rightarrow \text{BagEq}\ xs\ ys \rightarrow \text{Some}_0\ P\ xs \rightarrow \text{Some}_0\ P\ ys$$
$$\text{bag-as} \Rightarrow x s \cong y s \quad P x s = \text{let } (x, x \in x s, P x) = \text{find } P x s \text{ in}$$
$$\text{let } x \in y_s = \text{to } x_s \cong y_s \langle \$ \rangle_{x \in x_s}$$

in lose ($x, x \in y_s, P_x$)

```
module FindLoseCong {a la : Level} {A : Setoid a la} {P : A → SSetoid la la} {Q : A → SSetoid la la} where
```

open Membership A

open Setoid A

private

$$P_0 = \lambda e \rightarrow \text{Setoid.Carrier} (\Pi. \langle \$ \rangle \quad P e)$$
$$Q_0 = \lambda e \rightarrow \text{Setoid.Carrier } (\Pi. \langle \$ \rangle _ Q e)$$
$$\text{PSupport} = \lambda \text{ys} \rightarrow \sum y : \text{Carrier} \bullet y \in_0 \text{ys} \times \text{P}_0 y$$
$$\text{QSupport} = \lambda \text{ys} \rightarrow \sum y : \text{Carrier} \bullet y \in_0 \text{ys} \times \text{Q}_0 y$$
$$\approx : \{xs\ ys : List\ Carrier\} \rightarrow PSupport\ xs \rightarrow QSupport\ ys \rightarrow Set\ la$$
$$(a, a_{\text{exs}}, Pa) \approx (b, b_{\text{exs}}, Qb) = a \approx b \times a_{\text{exs}} \approx b_{\text{exs}}$$

open FindLose

```

find-cong : {ys : List Carrier} {p : Some0 P ys} {q : Some0 Q ys} → p ≈ q → find P p ≈ find Q q
find-cong (hereEq px qy) = refl , ≈-refl
find-cong (thereEq eq) = let (fst , snd) = find-cong eq in fst , thereEq snd

```

private

```

P+ : {x y : Carrier} → x ≈ y → P0 x → P0 y
P+ x≈y = Π. _ ($) _ ( _ ≈ _ .to (Π.cong P x≈y))

Q+ : {x y : Carrier} → x ≈ y → Q0 x → Q0 y
Q+ x≈y = Π. _ ($) _ ( _ ≈ _ .to (Π.cong Q x≈y))

lose-cong : {xs ys : List Carrier} {p : PSupport xs} {q : QSupport ys} → p ≈ q → lose P p ≈ lose Q q
lose-cong {p = a , here a≈x , Pa} {b , here b≈x , Qb} (fst , hereEq .a≈x .b≈x) = hereEq (P+ a≈x Pa) (Q+ b≈x Qb)
lose-cong {p = a , here a≈x , Pa} {b , there beys , Qb} (fst , ())
lose-cong {p = a , there a∈xs , Pa} {b , here px , Qb} (fst , ())
lose-cong {p = a , there a∈xs , Pa} {b , there beys , Qb} (a≈b , thereEq a∈xs≈beys) = thereEq (lose-cong (a≈b , a∈xs≈beys))

cong-fwd : {xs ys : List Carrier} {xs≈ys : BagEq xs ys} {p : Some0 P xs} {q : Some0 Q xs}
→ p ≈ q → bag-as⇒ P xs≈ys p ≈ bag-as⇒ Q xs≈ys q
cong-fwd {xs} {ys} {xs≈ys} {p} {q} p≈q with find P p | find Q q
...| (x , x∈xs , px) | (y , y∈ys , py) = lose-cong ({!need decidable equality?!} , {!!})

```

[Somebody: *Commented out:*

```

bag-as⇒ : {xs ys : List Carrier} → BagEq xs ys → Some0 P xs → Some0 P ys
bag-as⇒ xs≈ys Pxs = let (x , x∈xs , Px) = find Pxs in
  let x∈ys = to xs≈ys ($) x∈xs
  in lose (x , x∈ys , Px)

```

]

10.8 Some-cong and holes

This isn't quite the full-powered cong, but is all we need.

```

module _ {a ℓa : Level} {A : Setoid a ℓa} {P : A → SSetoid ℓa ℓa} where
open Membership A
open Setoid A
private

```

```

P0 = λ e → Setoid.Carrier (Π. _ ($) _ P e)
Support = λ ys → Σ y : Carrier • y ∈0 ys × P0 y

_ ≈ _ : {ys : List Carrier} → Support ys → Support ys → Set ℓa
(a , a∈xs , Pa) ≈ (b , b∈xs , Pb) = a ≈ b × a∈xs ≈ b∈xs
Σ-Setoid : (ys : List Carrier) → Setoid (ℓa ⊔ a) ℓa
Σ-Setoid ys = record
  {Carrier = Support ys
  ; _ ≈ _ = _ ≈ _
  ; isEquivalence = record
    {refl = λ {s} → Refl {s}
    ; sym = λ {s} {t} eq → Sym {s} {t} eq
    ; trans = λ {s} {t} {u} a b → Trans {s} {t} {u} a b
    }
  }
where
  Refl : Reflexive _ ≈ _
  Refl {a , a∈xs , Pa} = refl , ≈-refl
  Sym : Symmetric _ ≈ _

```

```

Sym (a≈b , a∈xs≈b∈xs) = sym a≈b , ≈-sym a∈xs≈b∈xs
Trans : Transitive _ ≈ _
Trans (a≈b , a∈xs≈b∈xs) (b≈c , b∈xs≈c∈xs) = trans a≈b b≈c , ≈-trans a∈xs≈b∈xs b∈xs≈c∈xs

module ≈ {ys} where open Setoid (Σ-Setoid ys) public
open FindLose P
open FindLoseCong hiding ( _ ≈ _ )
left-inv : {ys : List Carrier} (x∈ys : Some0 P ys) → lose (find x∈ys) ≈ x∈ys
left-inv (here px) = hereEq _ px
left-inv (there x∈ys) = thereEq (left-inv x∈ys)

right-inv : {ys : List Carrier} (pf : Σ y : Carrier • y ∈0 ys × P0 y) → find (lose pf) ≈ pf
right-inv (y , here px , Py) = (sym px) , (hereEq refl px)
right-inv (y , there y∈ys , Py) = (proj1 (right-inv (y , y∈ys , Py))) , (thereEq (proj2 (right-inv (y , y∈ys , Py))))

Σ-Some : (xs : List Carrier) → Some P xs ≅ Σ-Setoid xs
Σ-Some xs = record
  {to = record { _⟨$⟩_ = find {xs}; cong = find-cong }
  ;from = record { _⟨$⟩_ = lose; cong = lose-cong }
  ;inverse-of = record
    {left-inverse-of = left-inv
    ;right-inverse-of = right-inv
    }
  }

module _ {a ℓa : Level} {A : Setoid a ℓa} {P : A → SSetoid ℓa ℓa} where
open Membership A
open Setoid A
private P0 = λ e → Setoid.Carrier (Π. _⟨$⟩_ P e)
Some-cong : {xs1 xs2 : List Carrier} →
  (∀ {x} → (x ∈ xs1) ≅ (x ∈ xs2)) →
  Some P xs1 ≅ Some P xs2
Some-cong {xs1} {xs2} list-rel = record
  {to = record { _⟨$⟩_ = bag-as⇒ list-rel; cong = FindLoseCong.cong-fwd {P = P} {Q = P} }
  ;from = record { _⟨$⟩_ = xs1→xs2 (≅-sym list-rel); cong = {!!} }
  ;inverse-of = record {left-inverse-of = {!!}; right-inverse-of = {!!} }
  }
where
open FindLose P using (bag-as⇒; find)
  -- this is probably a specialized version of Respects.
  -- is also related to an uncurried version of 'lose'.
copy : ∀ {x} {ys} {Q : A → SSetoid ℓa ℓa} → x ∈0 ys → (Setoid.Carrier (Π. _⟨$⟩_ Q x)) → Some0 Q ys
copy {Q = Q} (here p) pf = here ( _≅_ .to (Π.cong Q p) ⟨$⟩ pf )
copy (there p) pf = there (copy p pf)

  -- this should be generalized to qy coming from Q0 x.
copy-cong : {x y : Carrier} {xs ys : List Carrier} {Q : A → SSetoid ℓa ℓa}
  (px : P0 x) (qy : Setoid.Carrier (Π. _⟨$⟩_ Q y)) (x∈xs : x ∈0 xs) (y∈ys : y ∈0 ys) →
  (x∈xs ≈ y∈ys) → copy {Q = P} x∈xs px ≈ copy {Q = Q} y∈ys qy
copy-cong px qy1 (here px1) ∘ (here qy) (hereEq .px1 qy) = hereEq _ _
copy-cong px qy (there i) ∘ (there _) (thereEq i≈j) = thereEq (copy-cong px qy _ _ i≈j)
xs1→xs2 : ∀ {xs ys} → (∀ {x} → (x ∈ xs) ≅ (x ∈ ys)) → Some0 P xs → Some0 P ys
xs1→xs2 {xs} rel p =
  let pos = find {ys = xs} p in
  copy ( _≅_ .to rel ⟨$⟩ proj1 (proj2 pos)) (proj2 (proj2 pos))
cong-fwd : {i j : Some0 P xs1} →
  i ≈ j → xs1→xs2 list-rel i ≈ xs1→xs2 list-rel j

```

$\text{cong-fwd } \{i\} \{j\} i \approx j = \text{copy-cong } \text{---} \{!!\}$

11 Conclusion and Outlook

???